

Sumario

Sumario	1
Formulación avanzada en Calc	3
Formulación avanzada	3
Objetivos	3
Objetivos	3
Habilidades previas	3
Versión imprimible del contenido	3
Sobre vídeos y capturas de pantalla	4
Uso del Asistente de funciones	4
El Asistente de funciones	4
Ficha Funciones	5
Buscar una función por su nombre	5
Buscar una función por categoría.	6
Localizar una función de la que desconocemos el nombre.	7
Otros elementos fundamentales del asistente de funciones	7
Insertar una función en la fórmula	8
Argumentos de la función	9
Seleccionar una celda o un rango de celdas como argumento	10
Ficha estructura	10
Ejemplo: insertar la función PROMEDIO con el asistente	11
Editar una fórmula existente con el asistente	13
Referencias relativas, mixtas y absolutas desde el asistente	13
La lista de funciones	13
Uso de nombres en fórmulas y funciones	14
Crear un nombre asignado a una celda o a un rango de celdas	15
Definir	16
Crear nombres desde la selección	17
Usar un nombre dentro de una fórmula	18
El operador ! de intersección	19
Modificar o eliminar un nombre asignado a una celda o rango	19
Uso de expresiones lógicas	20
Operadores lógicos y expresiones lógicas	20
Utilizar expresiones lógicas en funciones	22
Anidación de funciones	24
Indicar beneficios, resultado nulo o pérdidas	25
Está en plazo o ha vencido el plazo	28
Uso de comodines o expresiones regulares	33
Funciones de contar y sumar	34
Función CONTAR()	34
Función CONTARA()	35
Función CONTAR.SI()	36
Función CONTAR.SI() utilizando expresiones regulares	37
Función CONTAR.SI() utilizando criterios de comparación: =, >, <, >=, <=, <>	38
Función CONTAR.SI.CONJUNTO	39
Función SUMAR.SI	39
Función SUMAR.SI.CONJUNTO	41
Función CONTAR.BLANCO	42
Funciones de fecha y hora	42

Función HOY()	44
Función AHORA()	44
Función DIA()	45
Función MES()	45
Función AÑO()	46
Función FECHA()	46
Funciones de texto	47
Función CONCATENAR()	47
Funciones MAYUSC()	48
Funciones MINUSC()	48
Funciones NOMPROPIO()	48
Función LARGO()	49
Función HALLAR()	49
Función ENCONTRAR()	50
Funciones IZQUIERDA()	50
Funciones DERECHA()	51
Funciones MID()	51
Función REDUCIR()	51
Función TEXTO()	52
Funciones lógicas	53
Función SI()	53
Función NO()	55
Función Y()	55
Función O()	55
Funciones de hoja de cálculo	55
Función BUSCARV() y BUSCARH()	56
Función ÍNDICE()	59
Función COINCIDIR()	60
Practica lo aprendido	61
Ejercicio para practicar el uso del Asistente	62
Ejercicio de CONTAR.SI y SUMAR.SI	62
Ejercicio de cálculo con fechas	63
Ejercicios de funciones de texto	65
Ejercicios de funcion SI	66
Ejercicios de funciión BUSCARV	68
Ejercicio de referencias mixtas	70

Formulación avanzada en Calc

Formulación avanzada



Objetivos

En la presente unidad didáctica vamos a profundizar en técnicas de formulación, aprendiendo a explorar mediante el uso del **Asistente de funciones** las diferentes funciones que **Calc** pone a nuestra disposición, así como la creación y uso de **nombres** en las fórmulas y un amplio recorrido por las **funciones** de mayor utilidad.



Objetivos

Tras finalizar el estudio de esta unidad didáctica y realizar las prácticas sugeridas, habrás adquirido las habilidades de:

- Usar el Asistente de funciones para conocer las diferentes funciones que el programa nos ofrece.
- Crear nombres para referirnos a celdas y rangos de celdas, facilitando la formulación.
- Usar eficientemente una amplia colección de funciones para contar y sumar, manipular fechas y textos.
- Entender y usar las funciones anidadas.
- Entender la lógica de Calc y su aplicación a las fórmulas.



Habilidades previas

Para el correcto seguimiento de esta unidad didáctica es preciso tener un dominio razonable de las tareas relativas a:

- Edición y formateo básico de una hoja de cálculo
- Formulación básica con operadores aritméticos.
- Formulación básica con funciones (SUMA, PROMEDIO, CONTAR, MAX, MIN)



Versión imprimible del contenido

Desde estos enlaces se pueden descargar la versión imprimible de los contenidos y un archivo comprimido (.zip) con los materiales para la realización de las prácticas.

- [Versión imprimible de los contenidos \(Ventana nueva\)](#)



Sobre vídeos y capturas de pantalla

Los vídeos y capturas de pantalla de los contenidos de este curso han sido tomados con la **versión 5.3** de **LibreOffice** sobre un sistema operativo **Windows 7**. Si tienes instalada otra versión o ejecutas el programa en otra plataforma, puedes observar algunas diferencias. Además las elevadas posibilidades de personalización de la interfaz de usuario, pueden hacer que cambien el tamaño o apariencia de los iconos.

Algunas de las capturas de pantalla que ilustran los contenidos están reducidas; para verlas a mayor tamaño, basta pulsar sobre ellas para que se muestren en una ventana emergente. Pulsando **Esc** se cerrará la ventana emergente y se volverá al contenido.

Uso del Asistente de funciones

A estas alturas ya conocemos el uso de la función **SUMA** y otras funciones más habituales, como **MÁX**, **MÍN**, **PROMEDIO** o **CONTAR**.

Pero hay muchas más funciones que nos facilitarán la labor en nuestro trabajo con **LibreOffice Calc**.

En **Calc** existen cientos de funciones, muy orientadas a toda clase de tareas, y es realmente imposible conocerlas todas, por eso están organizadas en categorías: matemáticas, financieras, de texto, de fecha, etcétera.

Es responsabilidad del usuario intentar conocer las funciones que pueden ser más útiles para su trabajo.

Por ejemplo, un ingeniero de puentes utilizará especialmente las **funciones matemáticas y trigonométricas** que le ofrecerán un abanico de funciones que le permitirán calcular logaritmos, senos, cosenos o tangentes. Del mismo modo, quizás un contable no tenga ni idea de logaritmos, pero le pueden resultar muy útiles las diferentes **funciones financieras** que le ayudarán para el cálculo de préstamos, amortizaciones o cupones de bolsa. O un profesional de la estadística, hará buen uso de las **funciones estadísticas** que le permitirán evaluar desviaciones, distribuciones y tendencias.


El Asistente de funciones

El **Asistente de funciones** de **Calc** (en versiones anteriores denominado **AutoPiloto**) permite introducir funciones en celdas de modo interactivo.

Debemos activar el asistente de funciones en la celda donde deseamos insertar la función, y lo haremos por alguna de estas vías:

- Seleccionamos desde el menú **Insertar > Función**.
- Hacemos clic sobre el botón  **Asistente de funciones** en la barra de fórmulas.
- Pulsamos la combinación de teclas **Ctrl** + **F2**.

El asistente muestra el siguiente aspecto:

Inicio del **Asistente de funciones**


El diálogo presenta dos fichas llamadas **Funciones** y **Estructura**.

Ficha Funciones

Desde esta ficha podremos seleccionar las funciones que deseamos agregar a las fórmulas y completar sus argumentos.

Buscar una función por su nombre

La ficha **Funciones** incorpora (arriba a la izquierda) un cuadro de texto titulado **Buscar**. Escribiendo en este cuadro de texto un texto, la lista de funciones se filtrará por todas aquellas que su nombre comience, contenga o termine por este texto.




Buscando una función por parte de su nombre

Buscar una función por categoría.

La ficha **Funciones** dispone de un desplegable llamado **Categoría** que permite filtrar las funciones mostradas en la lista llamada **Función** situada justo debajo.


- Inicialmente tiene seleccionado el ítem **Todas**, por lo que en la lista situada justo debajo se muestran todas las funciones de que dispone **Calc**.
- También permite filtrar por una de las 11 categorías de funciones existentes (Texto, Estadística, Fecha y hora...).
- Finalmente, incluye en primera posición un ítem llamado **Usadas recientemente** que hace que se muestren sólo las funciones usadas más recientemente.

Funciones de la categoría **Lógico**

Localizar una función de la que desconocemos el nombre.

Para localizar la función deseada cuando no conocemos exactamente su nombre, podemos explorar las funciones filtrando por aquellas categorías que consideremos las más apropiadas; seleccionando con un clic el nombre de cada función en la lista situada justo debajo, se mostrará a su derecha el nombre de la función seleccionada, su sintaxis y una escueta descripción de la misma.

Con la función seleccionada podemos pulsar el botón **Ayuda**, que mostrará en una ventana emergente la ayuda ampliada de la función, que salvo algunas excepciones, presenta una descripción mucho más extensa y en muchas ocasiones incorpora también ejemplos de uso.

Descripción de la función **SI**

Otros elementos fundamentales del asistente de funciones

En la parte inferior del diálogo se muestra el cuadro llamado **Fórmula**; en éste se irá escribiendo la fórmula tal y como la vayamos completando con el asistente. También podemos editarla manualmente desde este cuadro; cualquier cambio introducido se refleja inmediatamente en la barra de fórmulas.

En la esquina superior derecha del diálogo se muestra un cuadro de texto llamado **Resultado de la función**. A medida que vamos completando el asistente, este cuadro muestra el resultado que se obtiene sólo evaluando la función actual.

Justo encima del cuadro **Fórmula**, y situado a la derecha, se muestra otro cuadro de texto llamado **Resultado**; a diferencia del anterior, éste muestra el resultado que se obtiene evaluando toda la fórmula introducida, y no sólo la función actual, ya que una fórmula puede contener más de una función.

Finalmente disponemos de un grupo de botones en donde destacamos **< Anterior y Siguiente >**, que nos permiten navegar entre las distintas pantallas del asistente, y si la fórmula contiene funciones anidadas, por los distintos niveles de anidamiento.

La casilla **Matriz** (en otras versiones, traducida incorrectamente como **Arreglo**) especifica que la fórmula insertada en el rango de celdas seleccionado será una fórmula de matriz.



Idea


Para conocer más sobre las fórmulas matriciales, puedes consultar el wiki de open-office.es en el siguiente enlace:

http://wiki.open-office.es/Formulas_matriciales_en_OpenOffice_Calc

Insertar una función en la fórmula

Para incorporar una función a la fórmula deberemos realizar estos pasos:

- Encontrar y seleccionar la función deseada
- Hacer un doble clic sobre su nombre para insertarla en la fórmula
- Si la función precisa de argumentos, definirlos desde los diferentes cuadros de texto que se muestran para cada uno de ellos.
 - Los argumentos son datos que debemos dar a la función para que pueda operar.
 - Los argumentos vienen precedidos por su nombre.
 - Al hacer clic sobre el cuadro de texto del argumento, se muestra su nombre y una información adicional sobre su uso dentro de la fórmula justo debajo de la descripción de la función.



Asistente para funciones con una fórmula completada

- Los argumentos pueden ser obligatorios u opcionales; a los segundos, si no se especifican, se les asigna un valor por defecto.

Según versiones, pueden mostrarse unos con su nombre en tipografía negrita, y los otros con tipografía corriente. En la versión actual se muestran todos con su nombre con tipografía corriente, si bien, siempre, tras el nombre mostrado en la descripción, se indica si el argumento es requerido o es obligatorio entre paréntesis.

Argumentos de la función

Los argumentos de una función son los datos que damos a la función para que haga su trabajo. Por lo tanto, el número y el tipo de argumento dependen de lo que requiere cada función.

Una función puede requerir argumentos de los siguientes tipos:

- Un valor constante: Teclearemos directamente el valor constante.
 - Si el argumento requiere un número, tecleamos el número sin incluir los puntos de los miles, pero sí la coma decimal si tiene decimales; por ejemplo **1500** o **1500000,50**.
 - Si el argumento requiere un texto, tecleamos el texto delimitado por comillas dobles, por ejemplo **"Valencia"**. Si no ponemos las comillas dobles, al hacer clic en cualquier otro cuadro Calc debe ponerlas.
 - Si argumento requiere una fecha o una hora podemos introducirlas como si de un texto se

tratase, por ejemplo, "31/12", o "31/12/17", o "10:30", o "10:30:28".

Si el documento es compartido por usuarios con otros formatos de fecha y hora, como por ejemplo, los anglófonos, puede ser problemático utilizar el formato escrito entre comillas dobles, por lo que recomendamos utilizar las funciones **FECHA()** y **TIEMPO()** para introducir estos valores.

- Una referencia a una celda o rango de celdas, relativa, absoluta o mixta, como por ejemplo **B15**, **\$B15**, **\$B\$15**, **B15:B45**. Es más recomendable hacer clic sobre la celda (o clic y arrastrar sobre el rango de celdas) antes que escribir la referencia de la celda, dado que se cometen menos errores. Son admitidos también nombres de celdas y nombres de rangos.
- Expresiones lógicas, como las palabras claves **VERDADERO** y **FALSO**, o pruebas lógicas, como **A2>B5**, **A2=2**, **A2<>"Valencia"**, **A2<="31/12/17"**, o **A2>=FECHA(2017;12;31)**.


Un argumento también puede venir dado por otra función, o por una función que opera con valores u otras funciones, como veremos en el apartado [Anidación de funciones](#). Para insertar una función dentro de un argumento utilizaremos el botón  **Insertar función**, situado a la izquierda del cuadro de texto del argumento.

Los valores lógicos **VERDADERO** y **FALSO** suelen ser sustituidos por el asistente por sus valores numéricos equivalentes, 1 para **VERDADERO**, 0 para **FALSO**.

Seleccionar una celda o un rango de celdas como argumento

Cuando precisamos seleccionar una celda o un rango de celdas como dato de un argumento, si el diálogo te oculta la zona de la hoja de cálculo que debes seleccionar, puedes desplazarlo a un lado desde su barra de título.

Al seleccionar la celda o rango de celdas el diálogo se reduce hasta ocupar sólo una línea.



Asistente encogido a una sola línea


Al soltar el botón del ratón el diálogo vuelve a su tamaño inicial.

Si no deseamos desplazar el asistente podemos temporalmente reducirlo haciendo clic sobre el botón  **Seleccionar** situado a la derecha del cuadro de texto del argumento. El diálogo se reducirá también hasta mostrar una sola línea

Seleccionaremos la celda o rango de celdas para a continuación hacer clic sobre el botón  **Expandir** para restaurar la ventana del asistente.

Ficha estructura

Esta ficha se utiliza para comprobar la estructura de las funciones anidadas y fórmulas complejas.




Estructura de funciones anidadas

Las operaciones aritméticas en la ficha **Estructura** se muestran en **notación polaca** (los operadores delante de los operandos), lo cual puede provocar cierta confusión si no estamos acostumbrados.

Ejemplo: insertar la función PROMEDIO con el asistente

Vamos a ver un ejemplo de cómo utilizar el **Asistente de funciones** para insertar la función **PROMEDIO** en una fórmula de una celda.

- Haremos clic en la celda donde deseamos que se muestre el promedio calculado
- Invocamos al **Asistente de funciones**.
- Seleccionamos la función PROMEDIO:
 - En el cuadro **Buscar** escribimos PRO; en la lista de funciones filtrada, veremos la función PROMEDIO; o..
 - En el desplegable **Categoría** seleccionaremos la categoría **Estadística**; en la lista de funciones filtrada, veremos la función PROMEDIO; o..
 - En desplegable **Categoría** seleccionaremos Todas; en la lista de funciones filtrada, hacemos clic en una función cualquiera, y tecleamos con suficiente rapidez **P R O**; en la lista se muestra iluminada la primera función que comienza por estas letras; fácilmente nos podemos desplazar hasta ver la función PROMEDIO.
- Con la función PROMEDIO seleccionada, haremos clic sobre el botón **Siguiente >**, o bien, haremos un doble clic sobre su nombre.

Ejemplo de uso del Asistente con la función **PROMEDIO**

- El asistente muestra información acerca de la sintaxis de la función y de los argumentos que precisa; vemos que **PROMEDIO** precisa argumentos numéricos, admite hasta un total de 30 como máximo, y el argumento llamado *número 1* se indica que es obligatorio.
- Observamos en el cuadro **Fórmula**, donde el asistente ya ha escrito la función, que está a la espera de que introduzcamos los argumentos.
- Si hacemos clic en los cuadros de texto de los argumentos **número 1**, **número2**, etc. la ayuda mostrada nos ofrece información sobre el argumento requerido. La barra de desplazamiento situada a la derecha de los argumentos nos permite navegar entre los mismos, cuando rebosan la capacidad de la ventana del diálogo.

**Advertencia**

Dado que todavía no le hemos especificado de qué datos tiene que calcular el promedio, y dado que los argumentos no tienen ningún valor por defecto, es normal que en **Resultado de la función** se muestre un mensaje de error **#DIV/0!**; también en **Resultado** de la fórmula se presenta un mensaje de error **#DIV/0!**. Por lo general, si una función presente en una fórmula ofrece como resultado un error, la fórmula también ofrecerá como resultado un error.

- También podemos hacer clic sobre el botón **Ayuda**. Una ventana emergente mostrará ayuda sobre las funciones de la categoría **Estadística**, entre las que se encuentra la función elegida.

- Una vez indicados los rangos de celdas de los que quieres calcular el promedio, haremos clic sobre el botón **Aceptar**. El resultado de la fórmula se mostrará en la celda.

Editar una fórmula existente con el asistente

Seleccionaremos la celda que contiene la fórmula, y llamaremos al asistente por cualquiera de los métodos anteriormente descritos; podremos editar la fórmula desde el mismo.

En caso de funciones anidadas, haciendo clic en cualquier parte de una función desde el cuadro **Fórmula**, se mostrarán los argumentos y la descripción de esa función y de sus argumentos.

Podremos utilizar también los botones **< Anterior** y **Siguiente >** para navegar entre funciones anidadas.

Referencias relativas, mixtas y absolutas desde el asistente

Al seleccionar la celda o rango de celdas que definen el argumento siempre se introducen como referencias relativas a la celda.

Cuando editamos una fórmula desde la barra de funciones o desde la propia celda haciendo un doble clic sobre ella, podemos utilizar la tecla de función **F4** para con cada pulsación convertirlas las referencias a celdas relativas en referencias absolutas, mixtas de fila, mixtas de columna, o de nuevo, relativas.

Lamentablemente el asistente no incorpora esta función, por lo que o bien introducimos los \$ manualmente, o tras finalizar la edición desde el asistente continuarla bien en la barra de fórmulas o en la propia celda.


La lista de funciones

Otra herramienta que puede resultar de gran ayuda para la creación de fórmulas es la **Lista de funciones**.

La Lista de funciones se puede activar:

- Desde el menú **Ver > Lista de funciones**.
- Desde el botón **Funciones** de la barra de herramientas lateral.

En cualquier caso se mostrará en el panel lateral una lista de funciones.



*Insertando la función **ALEATORIO.ENTRE** desde la **Lista de funciones***

En el desplegable superior del panel, pueden filtrarse las funciones por las mismas categorías que se muestran en el **Asistente de funciones**.

Haciendo clic sobre el nombre de una función, en la parte inferior se muestra una descripción breve junto a su sintaxis.

Si este área resulta demasiado pequeña, se puede ampliar arrastrando la divisoria entre la lista de funciones y el área de descripción.

Haciendo doble clic sobre un nombre de función, ésta se insertará en la celda activa, en modo de edición y con unos marcadores de posición para cada argumento obligatorio de la función. No se incluirán marcadores de posición para los argumentos opcionales.

Simplemente habrá que sustituir cada argumento, por el valor o referencia que le corresponda.

Uso de nombres en fórmulas y funciones

Cuando queremos hacer un cálculo con valores almacenados en celdas, utilizamos una fórmula con referencias (relativas, mixtas o absolutas) a las celdas, mediante su notación de columna y fila: **A1**, por

ejemplo.

Existe otro método para hacer referencia a una celda o a un rango de celdas. Consiste en aplicar un **nombre** a una celda o a un rango de celdas.


A partir de ese momento, podremos hacer referencia a esa celda o rango bien con su notación de fila y columna (el que hemos utilizado hasta ahora, por ejemplo **B5** o **H3:J14**) y también, de forma indistinta, por el **nombre** con el que los hemos denominado.

El uso de celdas o rangos con nombre tiene varias ventajas:

- Se pueden usar como **puntos o marcadores** a lugares específicos en hojas de cálculo grandes. Basta con desplegar el cuadro de nombres (a la izquierda de la barra de fórmulas) y seleccionar el nombre deseado para desplazar la celda activa a esa celda o rango, incluso aunque estemos trabajando en otra hoja dentro del libro.
- Mejoran la **legibilidad** de las fórmulas. Comparemos por ejemplo una expresión como **=SUMA(Hoja3.G5:G13)*PROMEDIO(Hoja2.A20:A25)** con esta otra: **=SUMA(VENTAS)*INFLACION_PREVISTA**.
- Una celda con nombre es una **referencia fija o absoluta**, mientras que un rango con nombre es una referencia mixta, por lo que al utilizarlo en una fórmula, no deberemos preocuparnos por añadir notaciones con el símbolo del dólar (\$columna\$fila).

Crear un nombre asignado a una celda o a un rango de celdas

Para crear un nombre, seleccionaremos la celda o rango de celdas a las que deseamos asignar un nombre, pulsaremos en el cuadro de nombres (a la izquierda de la barra de fórmulas) y escribiremos el nombre.



The screenshot shows the LibreOffice Calc interface. At the top, the 'Define Name' dialog box is open, with the text 'mes_enero' entered in the 'Name' field and '2954' in the 'Value' field. Below the dialog, a spreadsheet table is visible. The table has columns A, B, C, and D, and rows 1 through 8. The data in the table is as follows:

	A	B	C	D
1	VENTAS POR PROVINCIA PRIMER TRIMESTRE			
2				
3	Provincia	Enero	Febrero	Marzo
4	Alicante	2.954,00 €	2.885,00 €	1.744,00 €
5	Castellón	2.347,00 €	2.404,00 €	2.067,00 €
6	Valencia	1.804,00 €	2.925,00 €	2.017,00 €
7	Total	1.643,00 €	2.756,00 €	1.426,00 €
8				

Creando un nombre para un rango de celdas

Es muy importante finalizar la denominación pulsando la tecla **Intro**. De otro modo no funcionará.

Podemos verificar si el nombre ha sido correctamente creado desplegando el cuadro de nombres que mostrará todos los nombres creados. Seleccionando alguno de ellos, se activará la celda o rango correspondiente.

	B	C	D
	VENTAS POR PROVINCIA PRIMER TRIMESTRE		
3	Provincia	Enero	Febrero
4	Alicante	2.954,00 €	2.885,00 €
5	Castellón	2.347,00 €	2.404,00 €
6	Valencia	1.804,00 €	2.925,00 €
7	Total	1.643,00 €	2.756,00 €

Cuadro de nombres desplegado


Asimismo, si seleccionamos una celda o un rango de celdas con nombre, en lugar de mostrar la habitual referencia de columna y fila, el cuadro de nombres mostrará el nombre definidos para esa selección.

	A	B	C
1	VENTAS POR PROVINCIA PRIMER TRII		
2			
3	Provincia	Enero	Febrero
4	Alicante	2.954,00 €	2.885,0
5	Castellón	2.347,00 €	2.404,0
6	Valencia	1.804,00 €	2.925,0
7	Total	1.643,00 €	2.756,0
8			

Selección de un rango nombrado

Definir

Un método alternativo para asignar un nombre a una celda o rango de celdas, es desde el menú **Hoja > Intervalos y expresiones con nombre > Definir**, que mostrará el diálogo **Definir nombre**.



Diálogo **Definir nombre**

- En el cuadro **Nombre** escribe la denominación de la celda o rango.

- En **Intervalo** aparecerá la referencia a la selección de celdas; si no habías hecho la selección correcta a la que deseas asignar el nombre, lo puedes hacer desde aquí.
- Por defecto, los nombres se crean con **Alcance Documento (Global)**, lo que hará que el nombre esté disponible para todo el libro. El desplegable **Alcance** permite crear el nombre en el ámbito de una hoja determinada, lo que lo hará invisible a las demás hojas. Esto permitirá definir un mismo nombre de celda o rango de celdas en hojas diferentes.
- La sección expansible **Opciones del intervalo** permite asignar diferentes funcionalidades al rango cuyo nombre estás creando.



Advertencia.


Reglas sobre nombres

- **Caracteres válidos:** El primer carácter de un nombre debe ser una letra o un carácter de subrayado (_). El resto de los caracteres del nombre pueden ser letras, números, puntos y caracteres de subrayado. Podemos utilizar como nombre un sólo carácter, usando las letras a-z o A-Z, excepto las letras C c R r por compatibilidad con otras hojas de cálculo. Referencias a celdas no se permiten: Los nombres no pueden ser idénticos a una referencia de celda, como B15 o B\$15 o B5:C8.
- **No se admiten espacios:** Los espacios no están permitidos como parte de un nombre. Usa el carácter de subrayado (_) y el punto (.) como separadores de palabra, por ejemplo *Impuesto_Tipo_1* o *Impuesto.Tipo.1*, o utiliza combinaciones de mayúsculas y minúsculas, como *ImpuestoTipo1*.

Crear nombres desde la selección

Tenemos una posibilidad de crear varios nombres a la vez, seleccionando una tabla de datos y tomando como nombre las etiquetas de la primera fila o de la primera columna.

Tras realizar la selección, activaremos desde el menú **Intervalos y expresiones con nombre > Crear...** se mostrará el diálogo **Crear nombres**.



Crear nombres desde selección

Dado que en el ejemplo deseamos crear nombres con los encabezados del rango de celdas

seleccionados, marcamos las opciones **Fila superior** y **Columna izquierda**.

Con este solo paso, en el ejemplo mostrado se han creado los siguientes nombres:

Alicante	=Hoja1.\$C\$4:\$E\$4
Castellón	=Hoja1.\$C\$5:\$E\$5
Enero	=Hoja1.\$C\$4:\$C\$6
Febrero	=Hoja1.\$D\$4:\$D\$6
Marzo	=Hoja1.\$E\$4:\$E\$6
Provincia	=Hoja1.\$C\$4:\$E\$6
Valencia	=Hoja1.\$C\$6:\$E\$6

Usar un nombre dentro de una fórmula

Para usar dentro de una fórmula un nombre que ya hemos creado, podemos escribirlo directamente dentro de la fórmula.

Cuando en una fórmula comenzamos a escribir un nombre, una etiqueta emergente nos lo mostrará como pista para autocompletarlo pulsando **Intro**.

	A	B	C	D	E	F	G
1		VENTAS POR PROVINCIA PRIMER TRIMESTRE					
2							
3		Provincia	Enero	Febrero	Marzo		
4		Alicante	2.954,00 €	2.885,00 €	1.744,00 €	=SUMA(ALI	[Alicante]
5		Castellón	2.347,00 €	2.404,00 €	2.067,00 €		
6		Valencia	1.804,00 €	2.925,00 €	2.017,00 €		
7		Total	1.643,00 €	2.756,00 €	1.426,00 €		
8							

Nombre para autocompletar al escribir fórmula

Si no queremos escribir el nombre, mientras editamos la fórmula podemos acceder al menú **Insertar > Intervalo o expresión con nombre....**, desde donde se mostrará el diálogo **Pegar nombres**; pulsando el botón **Pegar**, el nombre quedará insertado en la fórmula.

	A	B	C	D	E	F	
1		VENTAS POR PROVINCIA PRIMER TRIMESTRE					
2							
3		Provincia	Enero	Febrero	Marzo		
4		Alicante	2.954,00 €	2.885,00 €	1.744,00 €	=SUMA(	
5		Castellón					
6		Valencia					
7		Total					
8							
9							
10							
11							
12							
13							
14							
15							
16							
17							

Pegar nombres

Nombre	Intervalo o expresión ...	Ámbito
Alicante	Hoja2.\$C\$4:\$E\$4	Documento (Global)
Castellón	Hoja2.\$C\$5:\$E\$5	Documento (Global)
Enero	Hoja2.\$C\$4:\$C\$6	Documento (Global)
Febrero	Hoja2.\$D\$4:\$D\$6	Documento (Global)
Marzo	Hoja2.\$E\$4:\$E\$6	Documento (Global)
mes_enero	Hoja2!\$C\$4:\$C\$6	Documento (Global)
mes_febrero	Hoja2!\$D\$4:\$D\$6	Documento (Global)
mes_marzo	Hoja2!\$E\$4:\$E\$6	Documento (Global)
Provincia	Hoja2.\$C\$4:\$E\$6	Documento (Global)

Ayuda

Pegar todo

Cerrar

Pegar

Si no estamos editando una fórmula, mediante el botón **Pegar todo** de este mismo diálogo, **Calc** insertará en la hoja una lista con los nombres definidos y las celdas o rangos de celdas correspondientes.

Alicante	=Hoja1.\$C\$4:\$E\$4
Castellón	=Hoja1.\$C\$5:\$E\$5
Enero	=Hoja1.\$I\$9
Febrero	=Hoja1.\$D\$4:\$D\$6
Marzo	=Hoja1.\$E\$4:\$E\$6
Mes_enero	=Hoja1.\$C\$4:\$C\$6
Provincia	=Hoja1.\$C\$4:\$E\$6
Valencia	=Hoja1.\$C\$6:\$E\$6

Listado de nombres creados

El operador ! de intersección


Cuando estamos trabajando con rangos nombrados, una alternativa a las referencias por columna y fila (A5, J10, etc) que podemos utilizar dentro de nuestras fórmulas es utilizar los nombres con el operador de intersección !.

Teniéndolo presente, será perfectamente válido escribir una fórmula como

=Febrero!Castellón+Marzo!Castellón

Modificar o eliminar un nombre asignado a una celda o rango

Para modificar o eliminar un nombre, desde el menú **Hoja > Intervalos y expresiones con nombre > Gestionar** (o pulsando **Ctrl + F3**) se mostrará el diálogo **Gestionar nombres** que te permitirá editar los nombres, modificar los rangos a que se refieren, o eliminarlos.

Diálogo **Gestionar nombres**

Pulsando el botón **Añadir** también se abrirá el diálogo **Definir nombre** visto anteriormente.

Uso de expresiones lógicas

Operadores lógicos y expresiones lógicas

Para introducirnos en la lógica de **Calc** en primer lugar analizaremos qué son las **expresiones lógicas** y conocer los **operadores lógicos**.

Hasta ahora, en nuestras fórmulas hemos usado fundamentalmente los **operadores aritméticos** (+ - / *).

Cuando creamos una expresión con operadores aritméticos estamos creando una **expresión aritmética**.

Pero también disponemos de otros operadores, denominados operadores de comparación, o en un término más preciso, **operadores lógicos**. Ya hemos usado operadores lógicos cuando aplicábamos un criterio numérico a una función **CONTAR.SI** o **SUMAR.SI**.

Relacionamos a continuación los operadores lógicos y sus descripciones.

- **Mayor que**, se representa con el símbolo **>** ; **x > y** se lee como x mayor que y

- **Mayor o igual que**, se representa con el símbolo $\geq$; $x \geq y$ se lee como x mayor o igual que y
- **Menor que**, se representa con el símbolo $<$; $x < y$ se lee como x menor que y
- **Menor o igual que**, se representa con el símbolo $\leq$; $x \leq y$ se lee como x menor o igual que y
- **Igual que**, se representa con el símbolo $=$; $x = y$ se lee como x igual que y
- **Distinto de**, se representa con el símbolo $\neq$; $x \neq y$ se lee como x distinto que y

Cuando creamos una expresión utilizando **operadores lógicos**, estamos creando una **expresión lógica**.

Como característica especial, las expresiones lógicas tan sólo devuelven dos resultados posibles: el valor lógico **VERDADERO** o el valor lógico **FALSO**.

Reproducimos los siguientes ejemplos:

- La fórmula $=5>3$ dará como resultado **VERDADERO**
- La fórmula $=5<3$ dará como resultado **FALSO**
- La fórmula $=\text{"a"}=\text{"A"}$ dará como resultado **VERDADERO**
- La fórmula $=\text{"a"}=\text{"á"}$ dará como resultado **FALSO**
- La fórmula $=\text{"z"}>\text{"a"}$ dará como resultado **VERDADERO**
- La fórmula $=\text{PI}()>3,14$ dará como resultado **VERDADERO**

Naturalmente, los ejemplos reales pueden ser bastante más complicados que los que se muestran aquí, como $=\text{SUMA}(\text{A1:A10})>\text{PROMEDIO}(\text{B1:B5})$.

Observemos algunas características que se deducen de los ejemplos mostrados:

- Las expresiones lógicas aplicadas a textos implican un sentido de ordenación: la "Z" es mayor que la "A".
- A estos efectos, mayúsculas y minúsculas son equivalentes.
- Caracteres con tilde o acento, no son equivalentes a los mismos caracteres sin ellos.

Gracias a las expresiones lógicas, podemos disfrutar de funciones como **SI()** que permiten obtener un resultado u otro en función de que una expresión lógica resulte como verdadera o falsa.

Existe una equivalencia numérica que permite crear operaciones aritméticas con expresiones lógicas. A estos efectos, **FALSO** equivale a cero y **VERDADERO** a uno o a cualquier valor distinto de cero.

Los valores lógicos **VERDADERO** o **FALSO** se pueden introducir directamente en fórmulas escribiéndolos tal cual, sin comillas, o bien como funciones **VERDADERO()** o **FALSO()**.

A la hora de generar expresiones lógicas en Calc lo más habitual es comparar una celda con:

- Otra celda. Ejemplos de fórmulas que compararan una celda con otra celda: $=\text{B2}=\text{B5}$, $=\text{B2}<\text{B5}$, $=\text{B2}>\text{B5}$, $=\text{B2}\neq\text{B5}$.
- Un valor constante numérico. Ejemplos de fórmulas que compararan una celda con un valor constante numérico: $=\text{B2}=2$, $=\text{B2}<5$, $=\text{B2}>1452,65$, $=\text{B2}\neq-25$.
- Un valor constante alfanumérico. Ejemplos de fórmulas que compararan una celda con un valor constante alfanumérico: $=\text{B2}=\text{"Valencia"}$, $=\text{B2}<\text{"B"}$, $=\text{B2}>\text{"Barcelona"}$, $=\text{B2}\neq\text{"Alicante"}$.
- Un valor constante fecha. Ejemplos de fórmulas que compararan una celda con un valor constante fecha: $=\text{B2}=\text{FECHA}(2017;12;31)$, $=\text{B2}>\text{"31/12/2017"}$.
- Un valor constante hora. Ejemplos de fórmulas que compararan una celda con un valor constante hora: $=\text{B2}=\text{TIEMPO}(18;15;30)$, $=\text{B2}<\text{"18:15:30"}$.

Utilizar expresiones lógicas en funciones

Algunas funciones realizan su función filtrando los datos mediante criterios de comparación (expresiones lógicas) en uno o más de sus argumentos.

Por ejemplo, veremos más adelante que **CONTAR.SI** se utiliza habitualmente con expresiones lógicas, y la utilizaremos si queremos contar en una lista de datos cuántos de ellos son iguales, mayores, mayores o iguales, menores, menores o iguales, o distintos a un valor o al contenido de una celda.

Por ello en uno de los argumentos de la función deberemos indicar el criterio (expresión lógica) que debe seguir; la forma correcta de escribir el criterio se resume en estos cuatro puntos:

- Si la condición es "igual a" un valor constante, se escribe el valor directamente dentro de la fórmula; en caso de que sea un número, sin comillas dobles; si es un texto debemos escribirlo entre comillas dobles siempre. Si el valor constante es una fecha u hora, recomendamos utilizar las funciones FECHA() y TIEMPO(), si bien también se acepta escribir la fecha como un texto con notación "AAAA-MM-DD", pero no en notación "DD/MM/AAAA". Si es igual a una celda, se inserta la referencia a la celda.
- **=CONTAR.SI(C2:C7;300)**: cuenta apariciones en el rango del valor 300 (igual a 300).
- **=CONTAR.SI(B2:B7;"Valencia")**: cuenta apariciones en el rango del valor "Valencia" (igual a Valencia).
- **=CONTAR.SI(R2:R7;FECHA(2017;12;31))**: cuenta apariciones en el rango del valor 31/12/2017 (igual a 31/12/2017).
- **=CONTAR.SI(R2:R7;"2017-12-31")**: cuenta apariciones en el rango del valor 31/12/2017 (igual a 31/12/2017).
- **=CONTAR.SI(R2:R7;"31/12/2017")**: cuenta apariciones en el rango del valor 31/12/2017 (igual a 31/12/2017).
- **=CONTAR.SI(B2:B7;J18)**: cuenta apariciones en el rango del valor de la celda J18 (igual a la celda J18).

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
10	Resultado:	1		
11	Fórmula:	=CONTAR.SI(C2:C7;300)		

- Si la condición es >, >=, <, <=, <> que un valor constante, se escribirá la condición completa entre comillas dobles.
- **=CONTAR.SI(C2:C7;"<=300")**: cuenta apariciones en el rango de valores menores o iguales a 300.
- **=CONTAR.SI(B2:B7;">=Barcelona")**: cuenta apariciones en el rango de valores mayores o iguales a "Barcelona", alfabéticamente, por lo que se incluirán Barcelona y Valencia, pero no Alicante.

Formulación avanzada en LibreOffice Calc

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
10	Resultado:	4		
11	Fórmula:	=CONTAR.SI(C2:C7;">=300")		

- Si la condición incluye un operador de comparación (es >, >=, <, <=, <> de un valor), y una referencia a una función o a una celda, se escribirá el operador entre comillas dobles, seguido del operador & y el resto de la condición, tal y como se indicó en el primer punto.

- =CONTAR.SI(C2:C7;"<=" & J16):** cuenta apariciones en el rango de valores menores o iguales al valor contenido en la celda J16.
- =CONTAR.SI(B2:B7;">=" & FECHA(2017;12;31)):** cuenta apariciones en el rango de valores mayores o iguales a 31/12/2017.

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	300		
10	Resultado:	4		
11	Fórmula:	=CONTAR.SI(C2:C7;">=" & B9)		

- Si la condición completa se escribe en una celda, se inserta la referencia a la celda.
- =CONTAR.SI(C2:C7;B9):** donde:
 - Si B9 contiene el texto ">300", cuenta apariciones en el rango de valores mayores a 300
 - Si B9 contiene el texto "<>300", contará aquellos distintos a 300
 - Si B9 contiene el número 300, contará aquellos iguales a 300.

Formulación avanzada en LibreOffice Calc

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	>=300		
10	Resultado:	3		
11	Fórmula:	=CONTAR.SI(C2:C7;B9)		


Anidación de funciones

Como ya sabemos, la mayoría de las funciones de **Calc** precisan argumentos.

Éstos pueden ser valores constantes (un número, una fecha, un texto...) o también referencias a una celda o un rango (B2, J4:I18) que pueden ser relativas, absolutas o mixtas, o incluso una celda o rango con nombre.

Pero a veces uno o más argumentos deberán ser calculados utilizando la misma u otra función.

Como si de unas matrioshkas (muñecas rusas) se tratase, nos encontraremos funciones que contienen en su interior como un argumento a otra función (a la que llamamos función anidada).



En el esquema mostrado (algo abstracto, pero que concreta la idea), la **FUNCION2**, a la que se le pasan dos argumentos, debe pasar su resultado como segundo argumento de la **FUNCION1**. El cuadro inferior del esquema muestra como habrá que formular el cálculo para que se comporte como se desea, reemplazando el segundo argumento **ARG_1B** por la **FUNCION2**.

La función anidada será resuelta por **Calc** en primera instancia, devolviendo su resultado a la función que la anida y que utilizará este valor como argumento.

A lo largo de los siguientes apartados veremos diferentes casos con funciones anidadas; para hacer boca, ahí van unos ejemplos...

Resultado	Fórmula
2014	=AÑO(HOY())
actual	=SI(B3=AÑO(HOY());"actual";SI(B3<AÑO(HOY());"pasado";"futuro"))
CINCO	=MAYÚSC(ELEGIR(5;"uno";"dos";"tres";"cuatro";"cinco"))
Fórmula	=INDIRECTO(DIRECCIÓN(2;3))

Indicar beneficios, resultado nulo o pérdidas

Supongamos que dados unos resultados económicos, deseamos que en una celda se muestre un texto indicando de si se han obtenido beneficios, pérdidas o un resultado nulo.

- Si el resultado es un valor mayor que cero, deberá poner BENEFICIOS
- Si el resultado es un valor menor que cero, deberá poner PÉRDIDAS
- Si el resultado es cero, deberá poner RESULTADO NULO


El resultado es un valor numérico que se encuentra en la celda C3 de nuestra hoja. El texto lo queremos mostrar en la celda B6.

	A	B	C
1		Ingresos	100.000,00 €
2		Gastos	87.500,00 €
3		Resultado	12.500,00 €
4			
5		El resultado ha sido	
6			
7			

Para resolver este ejemplo precisaremos del uso de la función **SI()**.

Seguiremos los siguientes pasos:

- Haremos clic en la celda **B6**.
- Invocaremos al asistente de funciones.
- Seleccionaremos la función **SI()** y haremos clic en el botón **Siguiente >**.



- Haremos clic en el primer argumento, Prueba_lógica, y haremos clic sobre la celda C3 para insertar su referencia; luego teclearemos >0.
- Haremos clic en el segundo argumento, Valor_si_verdadero, y escribiremos "BENEFICIO".

Asistente de funciones

Funciones Estructura

Buscar: si

Categoría: Lógico

Función:

- CAMBIAR
- FALSO
- NO
- O
- SI**
- SI.CONJUNTO
- SI.ERROR
- SI.ND
- VERDADERO
- XOR
- Y

SI

Resultado de la función: BENEFICIO

Especifica una prueba lógica que se desea efectuar.

Valor_si_verdadero (opcional): Es el valor que se devolverá si prueba_lógica es VERDADERO.

Prueba_lógica:  C3>0 

Valor_si_verdadero:  "BENEFICIO" 

Valor_si_falso:  


Fórmula: Resultado BENEFICIO

=SI(C3>0;"BENEFICIO")

☐ Matriz

Ayuda < Anterior Siguiente > Aceptar Cancelar

- Haremos clic en el tercer argumento, Valor_si_falso; ahora necesitamos insertar una nueva función SI() anidada como tercer argumento en la función anterior, para lo que haremos clic sobre el botón  insertar función.
- De nuevo daremos los pasos necesarios para seleccionar la función **SI()** y haremos clic en el botón **Siguiente >**.
- Haremos clic en el primer argumento, Prueba_lógica, y haremos clic sobre la celda C3 para insertar su referencia; luego teclearemos <0.
- Haremos clic en el segundo argumento, Valor_si_verdadero, y escribiremos "PÉRDIDA".
- Haremos clic en el tercer argumento, Valor_si_falso, y escribiremos "RESULTADO NULO".



- Haremos clic sobre el botón **Aceptar**. El supuesto ha sido resuelto.

Está en plazo o ha vencido el plazo

Disponemos de una lista de datos con unos documentos y su fecha de entrada.

Disponemos en las celdas D1:D3 de unos valores numéricos que definen el año, mes y día límite para considerar si estos documentos han llegado en plazo, o el plazo ya ha vencido.

Necesitamos una función **SI()** para que compare la fecha del documento con la fecha límite, y escriba en la celda de esa fila "ESTÁ EN PLAZO" si la fecha del documento es menor, o "HA VENCIDO EL PLAZO" en caso contrario.

Como este documento será enviado a un compañero inglés, debemos asegurarnos que la comparación de fechas se hace correctamente, por lo que debemos anidar la función **FECHA()** dentro de la función **SI()**.

Formulación avanzada en LibreOffice Calc

	A	B	C	D
1		Fecha límite	Año:	2017
2			Mes:	7
3			Día:	15
4				
5				
6		Fecha de entrada	Documento	Estado
7		14/04/17	DOC518500	
8		19/07/17	DOC542054	
9		07/03/17	DOC715015	
10		03/08/17	DOC813663	
11		28/08/17	DOC446509	
12		18/07/17	DOC240782	
13		17/07/17	DOC649915	

Seguiremos los siguientes pasos:

- Haremos clic en la celda D7.
- Invocaremos al asistente de funciones.
- Seleccionaremos la función **SI()** y haremos clic en el botón **Siguiente >**.

Asistente de funciones

Funciones | **Estructura**

Buscar:

Categoría: **Lógico**

Función:

- CAMBIAR
- FALSO
- NO
- O
- SI**
- SI.CONJUNTO
- SI.ERROR
- SI.ND
- VERDADERO
- XOR
- Y

SI Resultado de la función: Err:511

Especifica una prueba lógica que se desea efectuar.

Prueba_lógica (obligatorio)

Es cualquier valor o expresión que pueda evaluarse como VERDADERO o FALSO.

Prueba_lógica:

Valor_si_verdadero:

Valor_si_falso:

Fórmula:


Resultado: Err:511

☐ Matriz

Ayuda **< Anterior** **Siguiente >** **Aceptar** **Cancelar**

Formulación avanzada en LibreOffice Calc

- Haremos clic en el segundo argumento, Valor_si_verdadero, y escribiremos **"ESTÁ EN PLAZO"**.
- Haremos clic en el tercer argumento, Valor_si_falso, y escribiremos **"HA VENCIDO EL PLAZO"**.



- Haremos clic en el primer argumento, Prueba_lógica; ahora necesitamos insertar la función **FECHA()** anidada como primer argumento en la función anterior, para lo que haremos clic sobre el botón .
- Daremos los pasos necesarios para seleccionar la función **FECHA()** y haremos clic en el botón **Siguiente >**.
- Haremos clic en el primer argumento, año, y haremos clic sobre la celda **D1** para insertar su referencia.
- Haremos clic en el segundo argumento, mes, y haremos clic sobre la celda **D2** para insertar su referencia.
- Haremos clic en el tercer argumento, día, y haremos clic sobre la celda **D3** para insertar su referencia.

Asistente de funciones

Funciones **Estructura**

Buscar
FEC

Categoría
Fecha y hora

Función

- AHORA
- AÑO
- AÑOS
- DIA
- DIA.LAB
- DIA.LAB.INTL
- DIAS
- DIAS360
- DIAS.LAB
- DIAS.LAB_EXCEL2003
- DIAS.LAB.INTL
- DIASEM
- DIASENAÑO
- DIASENMES
- DOMINGOPASCUA
- ESANOBIESTO
- FECHA**
- FECHA.MES
- FECHADIF

FECHA Resultado de la función 15/07/17

Devuelve el número de serie de una fecha especificada.

día (obligatorio)
Un entero entre 1 y 31 que representa el día del mes.

año D1

mes D2


día D3

Fórmula Resultado ESTÁ EN PLAZO


=SI(FECHA(D1;D2;D3);"ESTÁ EN PLAZO";"HA VENCIDO EL PLAZO")

☐ Matriz Ayuda < Anterior Siguiente > Aceptar Cancelar

- Ahora necesitamos volver a la función **SI()** que contiene a esta función **FECHA()**; haremos clic en el botón **< Anterior** las veces necesarias para volver a la edición de esta función, o bien, haremos clic sobre cualquier parte de la función **SI** en el cuadro **Fórmula**.



- Haremos clic en el primer argumento, Prueba_lógica, justo delante de la función FECHA(); y haremos clic sobre la celda B7 para insertar su referencia; luego teclearemos <.




- Haremos clic sobre el botón **Aceptar**.

En la celda D7 se ha insertado la fórmula =SI(B7<FECHA(D1;D2;D3);"ESTÁ EN PLAZO";"HA VENCIDO EL PLAZO")

Cuando extendamos la fórmula a las celdas situadas más abajo, las referencias relativas cambiarán; como el desplazamiento es hacia abajo, cambian los números de filas, pero es un efecto que no deseamos excepto para la referencia B7; las otras referencias deben ser fijadas como referencias absolutas.

Por lo tanto, haremos clic sobre la celda D7 y en la barra de fórmulas editaremos la expresión, haciendo que las referencias a las celdas D1, D2 y D3 sean absolutas.



A continuación, extenderemos la fórmula al resto de celdas. El supuesto ha sido resuelto.

Uso de comodines o expresiones regulares

Las **expresiones regulares** son patrones avanzados de búsqueda que se usan habitualmente en los

diálogos de **Buscar y Reemplazar**.

Los **comodines**, más simples de utilizar, sólo están disponibles en las versiones más recientes de **LibreOffice**.

Tanto comodines como expresiones regulares pueden ser utilizados en las fórmulas, pero hay que optar por una de las dos posibilidades. Para ello desde el menú **Herramientas > Opciones > LibreOffice Calc > Calcular** encontramos tres botones de opción que nos permiten seleccionar entre:

- ☐ Permitir comodines en las fórmulas.
- ☐ Permitir expresiones regulares en las fórmulas.
- ☐ Ningún comodín ni expresión regular.

Por defecto, en las versiones más actuales del programa están activados por defecto los comodines.

En la siguiente tabla se muestra la equivalencia entre ambas opciones a la hora de utilizarlas.

Para encontrar ...	Comodines	Expresiones
Un carácter y necesariamente uno	? (interrogante de cierre)	. (punto)
Cualquier conjunto de caracteres o ninguno	* (asterisco)	. * (punto se

Y una pequeña muestra de ejemplos:

Para encontrar ...	Como
Pepe o Pepa , pero no Pep ni Pepiño (cuatro caracteres que comienzan por pep).	
Pep , Pepe , Pepa o Pepiño indistintamente (cualquier término que comienza por pep)	
Pep o Josep , pero no Pepe , Pepa o Pepiño (cualquier término que finaliza por pep)	
Pep , Josep , Pepe , Pepa , Pepiño o Impepinable (cualquier término que contiene pep , que incluye a los que comienzan y finalizan por pep)	

Aunque la potencia de las expresiones regulares puede ir mucho más allá de estos simples ejemplos, para la mayoría de casos en que sea necesario utilizar patrones de búsqueda, nos bastará con usar comodines. Por eso recomendamos mantenerlos activados como vienen por defecto.

Funciones de contar y sumar

Ya conocemos la funciones **SUMA** y **CONTAR**. Pero hay bastantes más funciones que nos permitirán contar celdas de una tabla que cumplan ciertas condiciones, contar celdas vacías, sumar valores de una tabla correspondientes a determinado criterio, etc.

Función CONTAR()

CONTAR permite contar las celdas con contenido numérico en un rango dado. Ignora las celdas sin contenido.

Sintaxis básica:

=CONTAR(rango_a_contar)

Formulación avanzada en LibreOffice Calc

En el ejemplo, se cuentan cuantos valores aparecen bajo la columna **Importe** en el rango de celdas C2 : C7. El resultado lo queremos mostrar en a celda B9, por lo que en ella insertaremos la fórmula:

=CONTAR (C2 : C7)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Resultado:	6		
10	Fórmula:	=CONTAR(C2:C7)		
11				

Pero la función **CONTAR** no sirve para celdas con contenido de texto, pues sólo cuenta valores numéricos. En este ejemplo contamos cuántos valores aparecen bajo la columna Nombre, en el rango de celdas A2 : A7.

El resultado lo queremos mostrar en a celda B9, por lo que en ella insertaremos la fórmula:

=CONTAR (A2 : A7)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Resultado:	0		
10	Fórmula:	=CONTAR(A2:A7)		
11				

Dado que en ese rango de celdas no hay ningún valor numérico (son nombres de personas) el resultado será cero.

Función CONTARA()

CONTARA cuenta las celdas con contenido (texto o número) en un rango dado. Ignora las celdas sin contenido.

Sintaxis básica:

=CONTARA (Rango_a_contar)

En el ejemplo, contaremos cuántas celdas contienen texto o números bajo la columna **Nombre**, en el rango de celdas A2 : A7.

El resultado lo queremos mostrar en a celda **B9**, por lo que en ella insertaremos la fórmula:

=CONTARA (A2 :A7)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Resultado:	6		
10	Fórmula:	=CONTARA(A2:A7)		
11				

Dado que en ese rango de celdas, todas ellas, aunque no contienen valor numérico, sí contienen textos, el resultado será 6.

En el siguiente ejemplo estamos calculando cuantas celdas tienen contenido bajo la columna **Estado**, en el rango de celdas **D2 :D7**.

El resultado lo queremos mostrar en a celda **B9**, por lo que en ella insertaremos la fórmula:

=CONTARA (D2 :D7)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Resultado:	4		
10	Fórmula:	=CONTARA(D2:D7)		
11				

El resultado es 4, ya que las celdas vacías no se cuentan, y existen cuatro celdas que contienen texto.

Función CONTAR.SI()

CONTAR.SI cuenta las celdas que cumplen un determinado criterio en un rango dado.

Sintaxis básica:

=CONTAR.SI (Rango_a_contar;Criterio)

donde:

- **Rango_a_contar** es la referencia al rango donde se cuenta
- **Criterio** una expresión de texto, numérica o con comodines. Es insensible a mayúsculas o

minúsculas.

En el ejemplo contaremos en el rango de celdas **D2:D7** situadas bajo el encabezado **Estado** cuántas contienen el texto **Pendiente**, que hemos escrito en la celda **B9**.

El resultado queremos mostrarlo en **B10**, por lo que en esa celda insertamos la siguiente fórmula:

=CONTAR.SI (D2:D7;B9)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	Pendiente		
10	Resultado:	2		
11	Fórmula:	=CONTAR.SI(D2:D7;B9)		
12				

Función CONTAR.SI() utilizando expresiones regulares

Un error común se produce cuando los usuarios al escribir una palabra en una celda, finalizan la edición con un espacio. Esos espacios extras pueden provocar que un criterio aparentemente correcto no funcione como se espera, ya que **Calc** considera como dos cosas completamente distintas los textos "Pendiente" sin espacio al final y "Pendiente " con un espacio al final.

Podemos optimizar el resultado usando [expresiones regulares en la fórmula](#) (comodines).

En el ejemplo siguiente contaremos en el rango de celdas **D2:D7** situadas bajo el encabezado **Estado** cuántas comienzan por el texto **Pend**, para lo que en la celda **B9** escribiremos la expresión regular **Pend.*** (que básicamente podemos traducir por celdas que comiencen por pend).

El resultado queremos mostrarlo en **B10**, por lo que en esa celda insertamos la siguiente fórmula:

=CONTAR.SI (D2:D7;B9)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	Pend.*		
10	Resultado:	2		
11	Fórmula:	=CONTAR.SI(D2:D7;B9)		
12				

Formulación avanzada en LibreOffice Calc

En este otro ejemplo, cuantas celdas bajo **Nombre**, en el rango de celdas **A2:A7** comienzan por **Pe** y contienen dos caracteres más.

La expresión regular la vamos a escribir en **B9**, donde tecleamos **Pe..** (nótese los dos puntos tras las dos letras).

El resultado será mostrado en **B10**, por lo que en esa celda incorporamos la siguiente fórmula:

=CONTAR.SI (A2:A7;B9)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	Pe..		
10	Resultado:	2		
11	Fórmula:	=CONTAR.SI(A2:A7;B9)		

El resultado es 2, pues encuentra Pepe y Pepa, pero no incluye a Pep, que sólo cuenta con una letra más tras **Pe**.

Función CONTAR.SI() utilizando criterios de comparación: =, >, <, >=, <=, <>

También podemos utilizar **CONTAR.SI** con expresiones numéricas y operadores de comparación (profundizaremos en ellos con el estudio de las funciones lógicas).

En el ejemplo se cuentan cuántos importes cumplen el criterio indicado en la celda **B9**, en este caso, cuántos son superiores a 300.

El resultado se muestra en la celda en **B10**, por lo que en esa celda incorporamos la siguiente fórmula:

=CONTAR.SI (C2:C7;B9)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	>300		
10	Resultado:	3		
11	Fórmula:	=CONTAR.SI(C2:C7;B9)		

Función CONTAR.SI.CONJUNTO

La función **CONTAR.SI** tiene la limitación de que sólo admite un criterio.

Podemos superar esa limitación utilizando la función **CONTAR.SI.CONJUNTO**, siempre y cuando todos los criterios sean de tipo Y. Esta función no se puede aplicar si los criterios son de tipo O o mezcla de ambos.

- Se entiende como criterios de tipo Y aquellos que para cumplirse, deben cumplirse todos: La ciudad es Valencia Y las ventas mayores de 300.
- Se entiende como criterios de tipo O aquellos que para cumplirse, debe cumplirse al menos uno de ellos: La ciudad es Valencia O las ventas mayores de 300.

CONTAR.SI.CONJUNTO cuenta en un rango de celdas cuántos cumplen uno o más criterios. Deben cumplirse todos los criterios (criterios de tipo Y).

Sintaxis básica:

=CONTAR.SI.CONJUNTO (Rango1;Criterio1; [Rango2] ; [Criterio2] ; ...)

donde

- **Rango1**, **Rango2**, etc son los rangos donde comprobar los respectivos criterios.
- **Criterio1**, **Criterio2**, etc son las expresiones de texto, numéricas o con comodines cuya coincidencias en **Rango1**, **Rango2**, etc, respectivamente, se contarán.
- Sólo es obligatorio el primer par de Rango/Criterio, siendo opcionales los sucesivos.

En el ejemplo se cuentan filas correspondientes a la ciudad de Valencia y cuyo importe es >300. Deben cumplirse ambas condiciones.

La ciudad, y el criterio de importe los tecleamos en **B9** y **B10**, respectivamente.

El resultado se muestra en **B11**, por lo que en esa celda incorporamos la siguiente fórmula:

=CONTAR.SI.CONJUNTO (C2:C7;B9;B2:B7;B10)

	A	B	C	D	E
1	Nombre	Ciudad	Importe	Estado	
2	Pepe	Barcelona	100		
3	Pepa		200	Pendiente	
4	Carmen	Valencia	300		
5	Joaquín		400	Pendiente	
6	Pepiño	Alicante	500	Cancelado	
7	Pep	Valencia	600	Cancelado	
8					
9	Criterio Importe:	>300			
10	Criterio Ciudad	Valencia			
11	Resultado:	1			
12	Fórmula:	=CONTAR.SI.CONJUNTO(C2:C7;B9;B2:B7;B10)			

Función SUMAR.SI

Formulación avanzada en LibreOffice Calc

¿Quién no tiene largas listas de datos con una columna para notas u observaciones? Muchas veces querríamos sumar importes que tengan una observación concreta. Con esta función podemos sumar valores que cumplan determinados criterios.

La función SUMAR.SI es similar a CONTAR.SI, pero incorpora un argumento más: qué celdas deseamos sumar.

SUMAR.SI suma valores de un rango que cumplan un criterio específico.

Sintaxis básica:

=SUMAR.SI(Rango_a_evaluar;Criterio;[Rango_a_sumar])

donde:

- **Rango_a_evaluar** es el rango donde se comprobará el criterio.
- **Criterio** es una expresión de texto, numérica o con comodines cuya coincidencia se comprobará en el **Rango_a_evaluar**.
- **Rango_a_sumar** es el rango que contiene los valores que se sumarán; es opcional y se puede omitir, si coincide con el **Rango_a_evaluar**.

En el ejemplo siguiente, se suman los importes superiores a 300. Como el criterio se comprueba en el propio rango de importes, no se precisa, por ser coincidente, el tercer argumento de la función.

El criterio de importe lo tecleamos en B9. El resultado se muestra en B10, por lo que en esa celda incorporamos la siguiente fórmula:

=SUMAR.SI (C2 : C7 ;B9)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	>300		
10	Resultado:	1500		
11	Fórmula:	=SUMAR.SI(C2:C7;B9)		

En el siguiente ejemplo se suman los valores bajo la columna **Importe** que en la columna **Estado** contengan cualquier expresión que comience por **Pend**

El criterio de estado lo tecleamos en B9. El resultado se muestra en B10, por lo que en esa celda incorporamos la siguiente fórmula:

=SUMAR.SI (D2 : D7 ;B9 ;C2 : C7)

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Criterio:	Pend.*		
10	Resultado:	600		
11	Fórmula:	=SUMAR.SI(D2:D7;B9;C2:C7)		
12				

Función SUMAR.SI.CONJUNTO

Las funciones **SUMAR.SI** y **CONTAR.SI** tienen la limitación de que sólo admiten un criterio.

La función **SUMAR.SI** tiene la limitación de que sólo admite un criterio.

Podemos superar esa limitación utilizando la función **SUMAR.SI.CONJUNTO**, siempre y cuando todos los criterios sean de tipo Y. Esta función no se puede aplicar si los criterios son de tipo O o mezcla de ambos.

- Se entiende como criterios de tipo Y aquellos que para cumplirse, deben cumplirse todos: La ciudad es Valencia Y las ventas mayores de 300.
- Se entiende como criterios de tipo O aquellos que para cumplirse, debe cumplirse al menos uno de ellos: La ciudad es Valencia O las ventas mayores de 300.

SUMAR.SI.CONJUNTO suma los valores de un rango cuyos correspondientes en uno o más rangos cumplen uno o más criterios. Deben cumplirse todos los criterios (criterios de tipo Y).

Sintaxis básica:

=SUMAR.SI.CONJUNTO(Rango_a_sumar;Rango1;Criterio1;[Rango2];[Criterio2];...)

donde

- **Rango_a_sumar** es el rango con los valores que se desean sumar.
- **Rango1**, **Rango2**, etc son los rangos donde comprobar los respectivos criterios.
- **Criterio1**, **Criterio2**, etc son las expresiones de texto, numéricas o con comodines cuya coincidencias en **Rango1**, **Rango2**, etc, respectivamente, se contarán.
- Sólo es obligatorio el primer par de Rango/Criterio, siendo opcionales los sucesivos.

En el ejemplo se suman los valores de Importe correspondientes a la ciudad de Valencia Y cuyo estado es Cancelado. Deben cumplirse ambas condiciones.

El criterio de estado y de ciudad los tecleamos en **B9** y **B10**, respectivamente.

El resultado se muestra en **B11**, por lo que en esa celda incorporamos la siguiente fórmula:

=SUMAR.SI.CONJUNTO(C2:C7;D2:D7;B9;B2:B7;B10)

	A	B	C	D	E
1	Nombre	Ciudad	Importe	Estado	
2	Pepe	Barcelona	100		
3	Pepa		200	Pendiente	
4	Carmen	Valencia	300		
5	Joaquín		400	Pendiente	
6	Pepiño	Alicante	500	Cancelado	
7	Pep	Valencia	600	Cancelado	
8					
9	Criterio Importe:	Cancelado			
10	Criterio Ciudad:	Valencia			
11	Resultado:	600			
12	Fórmula:	=SUMAR.SI.CONJUNTO(C2:C7;D2:D7;B9;B2:B7;B10)			
13					

Función CONTAR.BLANCO

Si necesitamos saber el número de celdas vacías en un rango lo haremos con esta función.

CONTAR.BLANCO cuenta las celdas vacías en un rango.

Sintaxis básica:

=CONTAR.BLANCO (Rango)

En el ejemplo se muestra cuantas filas tienen sin asignar la Ciudad.

	A	B	C	D
1	Nombre	Ciudad	Importe	Estado
2	Pepe	Barcelona	100	
3	Pepa		200	Pendiente
4	Carmen	Valencia	300	
5	Joaquín		400	Pendiente
6	Pepiño	Alicante	500	Cancelado
7	Pep	Valencia	600	Cancelado
8				
9	Resultado:	2		
10	Fórmula:	=CONTAR.BLANCO(B2:B7)		
11				

Funciones de fecha y hora

Calc dispone de abundantes funciones para manipular y hacer cálculos con fechas y horas.

Pero antes de verlas, será conveniente dejar claros algunos conceptos sobre fechas y horas que nos permitirán entender algunos resultados que nos podrían confundir.

En primer lugar, **una fecha es un número formateado de una manera especial**. Para comprobarlo, basta con escribir una fecha en una celda y cambiar su formato a número.

Formulación avanzada en LibreOffice Calc

	A	B
1	11/10/17	43.019
2		

Y ¿qué significa este número? Ni más ni menos que el número de días transcurridos desde el 30/12/1899 hasta la fecha que hemos escrito.

Podemos comprobarlo fácilmente escribiendo el número 0 en una celda y activar el formato numérico de fecha (con cuatro dígitos de año). Ésta se corresponde con la fecha base que se utiliza en Calc (salvo configuración en contra).

	A	B
1	28/12/99	-2
2	29/12/99	-1
3	30/12/99	0
4	31/12/99	1
5	01/01/00	2

Esta equivalencia entre fechas y números es la que permite hacer cálculos con las fechas, obteniendo cuantos días hay entre dos fechas, o que fecha resultará de incrementar el día actual con un determinado número de días.

Normalmente, **Calc** presupone el tipo de resultado que deseas obtener y lo mostrará en el formato adecuado. Pero en ocasiones podemos encontrar con que esperando obtener una fecha, realmente obtenemos un número, o viceversa. Cuando nos ocurre eso, la primera reacción es pensar que nos hemos equivocado al formular, pero antes de tomar otras medidas, cambiamos el formato numérico al tipo de dato que esperamos obtener. La mayoría de las veces se resuelve el problema.

Si necesitamos calcular el número de días transcurridos entre dos fechas, una simple resta será suficiente.

C1					
	A	B	C		
1	11/10/17	05/03/66	18.848		
2					

Análogamente, si deseamos obtener la fecha resultante de incrementar una fecha en un número de días, simplemente haremos una suma.

C2					
	A	B	C		
1	Fecha inicio	Días plazo	Vencimiento		
2	11/10/17	30	10/11/17		
3					

Por el mismo motivo, si un día equivale a la unidad, la hora es una fracción de la unidad. 24:00 horas equivalen a 1 día; 12:00 horas equivalen a 0,5 partes del día. 6:00 horas son 0,25 partes del día.

Recordemos que para introducir horas en una celda con formato de hora hay que escribirlas como HH:MM o HH:MM:SS

	A	B
1	00:00	0
2	06:00	0,25
3	12:00	0,5
4	18:00	0,75
5	00:00	1

Ese es el motivo por el cual, si deseamos multiplicar horas (en formato de hora) por un precio/hora, habrá además que **multiplicar por 24**.

E2					
	A	B	C	D	E
1	Hora inicio	Hora fin	Duración servicio	Precio/hora	Importa a abonar
2	09:00	12:00	03:00	10,00 €	30,00 €
3	15:30	17:15	01:45	10,00 €	17,50 €
4	18:30	21:45	03:15	12,50 €	40,63 €

Conozcamos ahora algunas funciones especialmente útiles al tratar con fechas y horas.

Las funciones que siguen son las que por regla general son consideradas más útiles por los usuarios. Recomendamos una exploración de las diferentes funciones de esta categoría, donde se pueden encontrar muchas otras que quizás tengan aplicación directa en sus tareas.

Función HOY()

HOY devuelve la fecha actual.

Sintaxis básica:

=HOY()

Esta función no tiene argumentos.

A1			
	A	B	C
1	11/10/17		

Función AHORA()

AHORA devuelve la fecha y hora actual.

Sintaxis básica:

=AHORA()

Esta función no tiene argumentos.

A1					
	A	B	C		
1	11/10/17				
2					



Idea.

Ambas funciones son dinámicas y actualiza el resultado mostrado (especialmente relevante para **AHORA**) cada vez que se recalcula la hoja de cálculo. La hoja de cálculo se recalcula (salvo que hayamos optado por el recálculo manual) cada que vez que modificamos los datos de alguna hoja. Si tenemos una función **AHORA** y queremos que se recalculé sin necesidad de modificar ningún dato, deberemos pulsar la tecla **F9**.



Idea

¿Necesitas introducir la fecha o la hora de hoy pero no deseas que se actualice? Prueba los siguientes trucos de teclado:

- **Ctrl** + **,** (símbolo coma) introducirá la fecha del día de forma fija.
- **Ctrl** + **Mayús** + **,** (símbolo coma) introducirá la fecha y hora de este momento, de forma fija.

Función DIA()

DÍA devuelve el número de día correspondiente a una fecha.

Sintaxis básica:

=DIA(Número_de_fecha)

B1					
	A	B			
1	11/10/17	11			
2					

Función MES()

MES devuelve el número de día correspondiente a una fecha.

Sintaxis básica:

=MES(Número_de_fecha)

B1					
	A	B			
1	11/10/17	10			
2					

Función AÑO()

AÑO devuelve el número de día correspondiente a una fecha. Sintaxis básica:

=AÑO(Número_de_fecha)

	A	B
1	11/10/17	2017
2		

Función FECHA()

FECHA permite construir una fecha a partir de un número de año, un número de mes y un número de día.

Sintaxis básica:

=FECHA (Año;Mes;Día)

donde

- **Año** es un número de año
- **Mes** es un número de mes
- **Día** es un número de día
- El orden de los argumentos se ha de mantener estrictamente.

	A	B	C
1	Día	11	
2	Mes	10	
3	Año	2017	
4	Resultado:	11/10/17	
5	Fórmula:	=FECHA(B3;B2;B1)	
6			

A la función **FECHA** se le pueden pasar argumentos para día y mes excedidos de los límites corrientes, como un día número 32 o un mes número 13. **Calc** sabe calcular esos casos y entiende que el día 32 (en un mes de 31 días) corresponde al día 1 del mes siguiente, y que el mes 13 corresponde a enero del año siguiente, como en el siguiente ejemplo.

	A	B	C
1	Día	32	
2	Mes	13	
3	Año	2017	
4	Resultado:	01/02/18	
5	Fórmula:	=FECHA(B3;B2;B1)	

Podemos usar la función **FECHA** para introducir una fecha dentro de una fórmula, pero dentro de las fórmulas también podemos escribir una fecha como un texto en notación "**AAAA-MM-DD**". Las dos fórmulas siguientes son perfectamente equivalentes:

=HOY () -FECHA (1958 ;12 ;3)

=HOY () -"1958-12-3"



Advertencia

Aunque también se admita introducir una fecha escrita como un texto con notación "DD/MM/AAAA" no es recomendable si hemos de intercambiar nuestra hoja con usuarios que utilicen otro formato de fechas.

Funciones de texto

En muchas ocasiones vamos a necesitar operar con textos, por ejemplo para unir unos textos con otros, o para descomponer un código de texto en diferentes partes, o efectuar conversiones entre mayúsculas y minúsculas, entre otras. Seguidamente vamos a detallar algunas de las que consideramos más utilizadas por los usuarios. Recomendamos, como siempre, revisar mediante el asistente para descubrir otras funciones disponibles.

Función CONCATENAR()

CONCATENAR une textos.

Sintaxis básica:

=CONCATENAR (Texto1 ; [Texto2] ; ...)

Donde **Texto1**, **Texto2**, etc... son textos o referencias a celdas con textos.

	A	B	C
1	Nombre:	Pepito	
2	Apellido:	Pérez	
3			
4	Resultado:	Pepito Pérez	
5	Fórmula:	=CONCATENAR(B1;" ";B2)	

Como alternativa a la función **CONCATENAR** podemos utilizar el operador **&** de concatenación.

Sintaxis básica:

=Texto1&[Texto2]&...

	A	B	C
1	Nombre:	Pepito	
2	Apellido:	Pérez	
3			
4	Resultado:	Pepito Pérez	
5	Fórmula:	=B1 & " " & B2	

El resultado final es idéntico.

Recuerda que los textos dentro de las fórmulas deben ir siempre entre comillas, por eso se ha introducido un espacio entrecomillado para separar el nombre del apellido en el resultado mostrado.

Funciones MAYUSC()

MAYÚSC convierte un texto a mayúsculas.

Sintaxis básica:

=MAYÚSC (Texto)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	EN UN LUGAR DE LA MANCHA
4	Fórmula:	=MAYUSC(B1)
5		



En versiones anteriores la función MAYUSC() se escribía con acento; en esta versión todas las funciones se escriben sin acento.

Funciones MINUSC()

MINÚSC convierte un texto a minúsculas.

Sintaxis básica:

=MINÚSC (Texto)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	en un lugar de la mancha
4	Fórmula:	=MINUSC(B1)
5		



En versiones anteriores la función MINUSC() se escribía con acento; en esta versión todas las funciones se escriben sin acento.

Funciones NOMPROPIO()

NOMPROPIO convierte a mayúscula la primera letra de cada palabra dentro de un texto.

Sintaxis básica:

=NOMPROPIO (Texto)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	En Un Lugar De La Mancha
4	Fórmula:	=NOMPROPIO(B1)
5		

Función LARGO()

LARGO obtiene la longitud en caracteres de una cadena de texto.

Sintaxis básica:

=LARGO(Texto)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	24
4	Fórmula:	=LARGO(B1)
5		

Función HALLAR()

HALLAR localiza la posición de un texto dentro de otro; la búsqueda es indiferente a mayúsculas o minúsculas.

Sintaxis básica:

=HALLAR(Texto_buscado;Texto_donde_buscar;[Posición_inicial])

donde

- **Texto_buscado** es el texto que se desea localizar dentro de **Texto_donde_buscar**
- **Posición_inicial** es un argumento opcional para indicar a partir de qué posición de **Texto_donde_buscar** se desea efectuar la búsqueda.

En el siguiente ejemplo, la función **HALLAR** encuentra la posición número 7, correspondiente a la 'l' de 'lugar' (aunque la búsqueda se hacía de una 'L' mayúscula).

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	7
4	Fórmula:	=HALLAR("L";B1)
5		

En este otro ejemplo, se repite la búsqueda pero a partir de la posición número 8, por lo que la función devuelve el resultado 16 que corresponde a la posición de la 'L' de 'La Mancha'.

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	16
4	Fórmula:	=HALLAR("L";B1;8)

Si falla la búsqueda (al no encontrar ninguna 'l' más allá de la posición 17), se devuelve el error **#VALOR**.

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	#¡VALOR!
4	Fórmula:	=HALLAR("L";B1;17)

Función ENCONTRAR()

ENCONTRAR localiza la posición de un texto dentro de otro; la búsqueda es sensible a mayúsculas o minúsculas.

Sintaxis básica:

=ENCONTRAR(Texto_buscado;Texto_donde_buscar;[Posición_inicial])

donde

- **Texto_buscado** es el texto que se desea localizar dentro de **Texto_donde_buscar**
- **Posición_inicial** es un argumento opcional para indicar a partir de qué posición de **Texto_donde_buscar** se desea efectuar la búsqueda.

En este ejemplo, como la búsqueda se está haciendo sobre la 'L' en mayúsculas, **ENCONTRAR** salta por alto la coincidencia de la 'l' minúscula de 'lugar' y devuelve la posición de la 'L' mayúscula de 'La Mancha'.

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	16
4	Fórmula:	=ENCONTRAR("L";B1)

Funciones IZQUIERDA()

IZQUIERDA obtiene un determinado número de caracteres por la izquierda de una cadena de texto.

Sintaxis básica:

=IZQUIERDA(Texto;Número_caracteres)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	En un lugar
4	Fórmula:	=IZQUIERDA(B1;11)

En este ejemplo, se extraen 11 caracteres de la izquierda del texto de referencia.

Funciones DERECHA()

DERECHA obtiene un determinado número de caracteres por la derecha de una cadena de texto.

Sintaxis básica:

=DERECHA(Texto;Número_caracteres)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	Mancha
4	Fórmula:	=DERECHA(B1;6)

En este ejemplo, se extraen 6 caracteres de la derecha del texto de referencia.

Funciones MID()

MID obtiene un determinado número de caracteres de una cadena de caracteres a partir de una posición inicial.

Sintaxis básica:

=MID(Texto;Posición_inicial;Número_caracteres)

	A	B
1	Texto:	En un lugar de La Mancha
2		
3	Resultado:	lugar
4	Fórmula:	=MID(B1;7;6)

En este ejemplo, se obtienen 6 caracteres a partir de la posición 7 del texto de referencial. Nótese que la cadena de texto devuelta incluye un espacio inapreciable, que sería el sexto carácter de la cadena obtenida.

Función REDUCIR()

Esta función, equivalente a la función **ESPACIOS** de otros programas, elimina los espacios iniciales, finales y deja tan sólo un espacio de separación entre palabras de haber más de uno.

REDUCIR elimina espacios sobrantes de un texto.

Sintaxis básica:

=REDUCIR(Texto)

Formulación avanzada en LibreOffice Calc

	A	B	C
1			LARGO
2	Texto:	En un lugar de La Mancha	33
3			
4	Resultado:	En un lugar de La Mancha	24
5	Fórmula:	=REDUCIR(B2)	

En el ejemplo se escribió con espacios adicionales antes, después y dentro del texto. Se ha subrayado el texto para que se aprecien los espacios adicionales incluidos.

Se muestra el resultado de **LARGO** como comprobación, antes y después de aplicar la función **REDUCIR**.

Función TEXTO()

Esta función permite convertir un valor numérico en un texto representando el mismo valor con un formato determinado. Resultan muy útiles en aquellos casos en que no basta con cambiar el formato de un número para obtener la información deseada.

Por ejemplo, podríamos mostrar una fecha en un formato que sólo muestre el nombre del mes, pero realmente la celda seguiría conteniendo la fecha. Usando esta función obtendremos realmente un texto con el nombre del mes. También puede ser muy útil cuando necesitemos concatenar textos con números.

TEXTO convierte un valor numérico en un texto que representa el valor en un formato dado.

Sintaxis básica

=TEXTO(Valor;Código_formato)

	A	B
1	Valor:	43020
2	Código formato:	dddd dd \d\e mmmm \d\e aaaa
3		
4	Resultado:	jueves 12 de octubre de 2017
5	Fórmula:	=TEXTO(B1;B2)

donde **Valor** es un importe numérico y **Código_formato** uno de los formatos que se muestran en el cuadro **Código de formato** en la pestaña **Números** del cuadro de diálogo **Formato Celdas** cuando se aplica un formato de número.

Algunos ejemplos de código de formato aplicables:

	A	B
20	Importe:	42100
21	Código formato:	Resultado
22	d	6
23	dd	06
24	ddd	lun
25	dddd	lunes
26	m	4
27	mm	04
28	mmm	abr
29	mmmm	abril
30	000000	042100
31	#.###,00	42.100,00
32	aaaa	2015
33		

Una búsqueda de **Códigos de formato numérico** en la Ayuda del programa mostrará más información sobre como construir formatos numéricos personalizados, que podemos utilizar en el diálogo de **Formato celdas** así como con la función **TEXTO**.



Advertencia

Si deseamos insertar un texto literal debemos anteceder cada caracter con una barra invertida \ (código de escape) para que el formato no lo interprete como un código.

Funciones lógicas

En diferentes ocasiones podemos vernos en la necesidad de evaluar simultáneamente diferentes condiciones lógicas, para comprobar si se cumplen todas o sólo alguna de ellas, así como también comprobar la negación de una expresión lógica.

Para ello disponemos de estas funciones **NO()**, **Y()**, **O()**.

O tomar decisiones, evaluando una pregunta y actuando de una forma si la respuesta es **VERDADERO**, o de otra si es **FALSO**. Para tomar decisiones disponemos de la función **SI()**.

Función SI()

SI permite obtener un valor u otro en función de una expresión lógica dada.

Sintaxis básica:

=SI (Expresión_lógica; [Valor_si_VERDADERO] ; [Valor_si_FALSO])

donde:

- **Expresión_lógica** es la condición o comparación que se desea evaluar
- **Valor_si_VERDADERO** es el resultado que devolverá la función si la Expresión_lógica se resuelve como **VERDADERO**. Este argumento es opcional; de omitirlo, la función devolverá **VERDADERO**.
- **Valor_si_FALSO** es el resultado que devolverá la función si la Expresión_lógica se resuelve como

En los siguientes ejemplos se muestra el texto APROBADO si la nota del examen es igual o superior a 5 o SUSPENDIDO en caso contrario.

	A	B	C	D
13	Nota examen	Calificación		
14	3	=SI(A14>=5;"APROBADO";"SUSPENDIDO")		
15	5	APROBADO		
16	4	SUSPENDIDO		
17	6	APROBADO		
18	8	APROBADO		

La función **SI()** tiene una limitación y es que al ser tan binaria, para ella somos bajos o altos, gordos o delgados, blancos o negros, pues sólo nos da a escoger entre dos opciones. Y la vida real tiene muchos más matices. Veamos un ejemplo tan sencillo como el anterior, en que a continuación de unos resultados económicos se muestra el término BENEFICIOS o PÉRDIDAS en función de los mismos.
¿Qué ocurre cuando el resultado es exactamente cero?

	A	B	C	D
20	Resultados	Calificación		
21	1.253,00 €	=SI(A21>0;"BENEFICIOS";"PÉRDIDAS")		
22	23.456,00 €	BENEFICIOS		
23	-123,00 €	PÉRDIDAS		
24	-54.522,00 €	PÉRDIDAS		
25	0,00 €	PÉRDIDAS		

A todos nos consta que un resultado cero no representa beneficios de ningún tipo, pero tampoco podemos afirmar que se trate de pérdidas. En un caso así, necesitaríamos una tercera opción, NULO o algo parecido para poder calificar correctamente. Ahí es donde entra en juego el anidamiento de funciones. Después de comprobar que el resultado no es mayor que cero, quedan dos posibilidades: que sea igual a cero o que sea menor que cero. Por consiguiente, sustituiremos el tercer argumento de la función **SI()**, por otra función **SI()** que evalúe esta segunda expresión. En el caso del ejemplo...

	A	B	C	D	E
20	Resultados	Calificación			
21	1.253,00 €	=SI(A21>0;"BENEFICIOS";SI(A21=0;"NULO";"PÉRDIDAS"))			
22	23.456,00 €	BENEFICIOS			
23	-123,00 €	PÉRDIDAS			
24	-54.522,00 €	PÉRDIDAS			
25	0,00 €	NULO			

Hemos sombreado la función anidada para facilitar su lectura.

Por tanto, podemos deducir que para evaluar dos opciones, basta con una función **SI()**, para evaluar más opciones, necesitaremos una nueva función **SI()** por cada nueva opción.

El incremento de funciones anidadas puede convertir la fórmula en algo inmanejable, proclive a errores difíciles de depurar. Por eso, si vemos que necesitamos un excesivo número de funciones **SI** anidadas, comprobaremos en el próximo apartado que, con una pequeña modificación de enfoque, podemos resolver el problema con extrema sencillez usando la función **BUSCARV**.

Función NO()

NO invierte el valor lógico del argumento.

Sintaxis básica:

=NO(Expresión_lógica)

	A	B
28	Fórmula:	=NO(VERDADERO())
29	Resultado:	FALSO
30		

Función Y()

Y devuelve **VERDADERO**, si todos los argumentos son **VERDADERO**; en cualquier otro caso, devuelve **FALSO**.

Sintaxis básica:

=Y(Expr_lógica1; [Expr_lógica2]; ...)

	A	B	C
32	Y	VERDADERO	FALSO
33	VERDADERO	VERDADERO	FALSO
34	FALSO	FALSO	FALSO

En el ejemplo, una *tabla de la verdad* de la función **Y**.

Función O()

O devuelve **FALSO**, si todos los argumentos son **FALSO**; en cualquier otro caso, devuelve **VERDADERO**.

Sintaxis básica:

=O(Expr_lógica1; [Expr_lógica2]; ...)

	A	B	C
32	O	VERDADERO	FALSO
33	VERDADERO	VERDADERO	VERDADERO
34	FALSO	VERDADERO	FALSO

En el ejemplo, una *tabla de la verdad* de la función **O**.

Funciones de hoja de cálculo

Existe en **Calc** una categoría de funciones denominada **funciones de hoja de cálculo**, que ofrecen un importante conjunto de funciones para obtener referencias e información sobre la ubicación y contenido

de celdas, rangos, filas o columnas dentro de la hoja de cálculo.

En otros programas de hoja de cálculo, esta categoría de funciones se denomina como **funciones de búsqueda y referencia**.

Función BUSCARV() y BUSCARH()

La función **BUSCARV** (V por Vertical) nos permite buscar un valor en la primera columna de una tabla de datos y obtener el contenido de otra columna de la tabla en la misma fila del valor buscado.

La función **BUSCARH** (H por horizontal) funciona de forma análoga pero en tablas orientadas en filas en vez de en columnas. Es decir, se hace la búsqueda en la primera fila y se obtiene el valor de otra fila en la misma columna.

En ambos casos, podemos elegir entre que la búsqueda sea exacta o aproximada, en cuyo caso se exige que esté ordenada la primera columna o fila, según se trate de **BUSCARV** o **BUSCARH**.

Vamos a detallar el funcionamiento de **BUSCARV**, teniendo presente que la información ofrecida es perfectamente aplicable a **BUSCARH** siempre que tengamos los datos en filas.

BUSCARV busca un valor específico en la primera columna y devuelve el valor correspondiente a la misma fila desde otra columna.

Sintaxis básica:

=BUSCARV(Dato_buscado;Rango_donde_buscar;Número_columna;Búsqueda_aproximada)

donde:

- **Dato_buscado** es el valor que se desea localizar en la primera columna del **Rango_donde_buscar**.
- **Rango_donde_buscar** es la tabla de datos en cuya primera columna se efectuará la búsqueda.
- **Número_columna** es el número de columna del rango que contiene el valor que se desea obtener (es el número de columna dentro del rango, indiferentemente de que se trate de la columna **A**, la **Z** o cualquier otra)
- **Búsqueda_aproximada** es un argumento opcional; si se omite o es **1** o **VERDADERO**, la búsqueda será aproximada y se ubicará la fila que contenga el mayor valor menor al buscado, lo cual exige la la tabla esté ordenada por su primera columna; si se especifica **0** o **FALSO** la búsqueda será exacta y si no tiene éxito devolverá el valor de error **#N/D** (No disponible),

Hemos comentado al tratar la función **SI** que un exceso de funciones **SI** anidadas podía reemplazarse por una función **BUSCARV**. Veamos un ejemplo del caso.

Se trata de obtener las calificaciones de unos alumnos en función de su nota final. En el rango **E7:F10** se ha establecido una tabla con los diferentes tramos de calificación (se han añadido títulos para facilitar su comprensión pero no son relevantes a efectos de la fórmula). He aquí la solución,

Formulación avanzada en LibreOffice Calc

	A	B	C	D	E	F
1	Nota final	con SI	con BUSCARV			
2	7	=SI(A2<\$E\$8;\$F\$7;SI(A2<\$E\$9;\$F\$8;SI(A2<\$E\$10;\$F\$9;\$F\$10)))				
3	5	Aprobado				
4	7	Notable				
5	3	Suspense			Tramos calificación	
6	1	Suspense			Desde	Calificación
7	10	Excelente			0	Suspense
8	8	Notable			5	Aprobado
9	7	Notable			7	Notable
10	1	Suspense			9	Excelente

Dada la complejidad de la fórmula vamos a detallar su comportamiento. Con el primer **SI** comprobamos si la nota es inferior a **5**, en cuyo caso se determina que es un **Suspense**; si no, en el segundo **SI** comprobamos si la nota es inferior a **7**, en cuyo caso se trata de un **Aprobado**; si no se cumple ninguna de las anteriores, todavía hay dos posibilidades, que sea **Notable** o **Excelente**, por lo que volvemos a utilizar un tercer **SI** que comprueba si la nota es menor a **9**, en cuyo caso se trata de un **Notable**; como ya no quedan más posibilidades, no hace falta efectuar ninguna comprobación más, pues si no es **Suspense**, ni **Aprobado**, ni **Notable**, tan sólo puede ser **Excelente**.

La fórmula resultaría más legible si se hubiesen escrito directamente las notas de corte y calificaciones dentro de la misma, pero si en un momento determinado se decide que en vez de calificar como Suspense se califique como Insuficiente, o que la nota de corte del aprobado sea el 6 en vez del 5, no tendríamos más remedio que reescribir la fórmula. Tal como lo hemos hecho, cambiando los valores en la tabla de tramos de calificación, la fórmula seguirá funcionando a la perfección.

Es comprensible pensar que es muy fácil cometer errores al escribir una fórmula como ésta, y que además resultará difícil localizarlos.

Veamos como resolver el mismo problema con la función **BUSCARV**.

	A	B	C	D	E	F
1	Nota final	con SI	con BUSCARV			
2	7	Notable	=BUSCARV(A2;\$E\$7:\$F\$10;2)			
3	5	Aprobado	Aprobado			
4	7	Notable	Notable			
5	3	Suspense	Suspense		Tramos calificación	
6	1	Suspense	Suspense		Desde	Calificación
7	10	Excelente	Excelente		0	Suspense
8	8	Notable	Notable		5	Aprobado
9	7	Notable	Notable		7	Notable
10	1	Suspense	Suspense		9	Excelente

Podemos comprobar la simplicidad y elegancia de esta otra solución. **BUSCARV** busca la nota dentro de la primera columna del rango con los tramos de calificación y devuelve el contenido de la columna 2 del rango. Al tratarse de una búsqueda aproximada (hemos omitido el cuarto argumento), si no encuentra la nota, se queda en el tramo anterior; por ejemplo buscando la nota 3, como no la encuentra se queda en el tramo del 0, si busca un 8 se queda en el tramo del 7, etc.

Para este caso nos ha sido muy útil el uso de la búsqueda aproximada. También sería el caso de crear un modelo de hoja de cálculo para, por ejemplo, la declaración de la renta, cuya base liquidable está organizada por tramos.

Pero no siempre será lo más apropiado. Imaginemos un sistema de facturación donde hemos de emitir una factura al cliente 300 y al no encontrarlo, **Calc** le emitiese la factura al cliente 290. Para casos así,

Formulación avanzada en LibreOffice Calc

necesitaremos el uso del cuarto argumento para que el comportamiento de **BUSCARV** sea el correspondiente al de una búsqueda exacta.

Veamos el siguiente ejemplo. Se trata de una pequeña lista de datos y un formulario. En el formulario, al introducir un **Id** de contacto en la celda sombreada en amarillo, se debe mostrar correctamente el **Nombre**, **Ciudad** e **Importe** correspondiente a ese **Id**.

En principio, usando el **BUSCARV** como habíamos hecho hasta ahora, funciona aparentemente bien.

	A	B	C	D	E	F	G	H	I
12	DATOS					FORMULARIO			
13	Id	Nombre	Ciudad	Importe		Id	3		
14	1	Joaquín	Alicante	100		Nombre	=BUSCARV(\$G\$13;\$A\$14:\$D\$18;2)		
15	3	Carmen	Valencia	200		Ciudad	Valencia		
16	5	Pepito	Barcelona	300		Importe	200		
17	7	Fulano	Madrid	400					
18	9	Mengana	Reus	500					

Pero si introducimos un **Id** inexistente, nos mostrará un dato incorrecto; en este caso al no encontrar el **Id** 8, nos muestra los datos correspondientes al **Id** 7.

	A	B	C	D	E	F	G	H	I
12	DATOS					FORMULARIO			
13	Id	Nombre	Ciudad	Importe		Id	8		
14	1	Joaquín	Alicante	100		Nombre	Fulano		
15	3	Carmen	Valencia	200		Ciudad	=BUSCARV(\$G\$13;\$A\$14:\$D\$18;3)		
16	5	Pepito	Barcelona	300		Importe	400		
17	7	Fulano	Madrid	400					
18	9	Mengana	Reus	500					

Modifiquemos la fórmula añadiendo un cuarto argumento **0** a la función **BUSCARV** y nos mostrará el error **#N/D** advirtiéndole de la inexistencia del **Id** 8.

	A	B	C	D	E	F	G	H	I
12	DATOS					FORMULARIO			
13	Id	Nombre	Ciudad	Importe		Id	8		
14	1	Joaquín	Alicante	100		Nombre	#N/D		
15	3	Carmen	Valencia	200		Ciudad	=BUSCARV(\$G\$13;\$A\$14:\$D\$18;3;0)		
16	5	Pepito	Barcelona	300		Importe	#N/D		
17	7	Fulano	Madrid	400					
18	9	Mengana	Reus	500					

Naturalmente, si el **Id** introducido es correcto, la fórmula mostrará los datos correctos.

	A	B	C	D	E	F	G	H	I
12	DATOS					FORMULARIO			
13	Id	Nombre	Ciudad	Importe		Id	5		
14	1	Joaquín	Alicante	100		Nombre	=BUSCARV(\$G\$13;\$A\$14:\$D\$18;2;0)		
15	3	Carmen	Valencia	200		Ciudad	Barcelona		
16	5	Pepito	Barcelona	300		Importe	300		
17	7	Fulano	Madrid	400					
18	9	Mengana	Reus	500					



El caso del ejemplo funcionará correctamente aunque los datos estén en un libro diferente, ¡incluso cerrado!.

Podemos comprobarlo realizando los siguientes pasos:

1. Reproduciremos en un documento nuevo el ejemplo mostrado en una hoja denominada DATOS.
2. Moveremos los datos del formulario a una hoja diferente dentro del mismo documento denominada FORMULARIO.
3. Moveremos la hoja FORMULARIO a un libro nuevo.
4. Guardaremos el documento con la hoja DATOS y lo cerraremos.
5. En el libro con la hoja FORMULARIO, probaremos diferentes valores para la celda con el Id.

Observaremos que la fórmula sigue funcionando a la perfección.

Hemos visto que **BUSCARV** se busca en la primera columna (o en la primera fila si se trata de **BUSCARH**), pero ¿qué ocurre si el dato que necesitamos buscar está en otra columna?

Por ejemplo, si en el ejemplo anterior, hemos de buscar un **Nombre** para encontrar su **Id**.

Con lo que conocemos hasta el momento no tendríamos más remedio que modificar la tabla cambiando el orden de las columnas.

Pero eso no será preciso gracias a otras **funciones de hoja de cálculo** como **ÍNDICE** y **COINCIDIR**. Vamos a explorarlas.

Función ÍNDICE()

ÍNDICE devuelve una referencia a una celda o un intervalo dentro de un rango en base a su posición de fila o columna.

Sintaxis básica:

=ÍNDICE (Rango;Número_fila;Número_columna;Número_área)

donde

- **Rango** es el rango de referencia, que puede ser múltiple, es decir, compuesto por varias áreas, en cuyo caso deberá ir entre paréntesis.
- **Número_fila** es un argumento opcional que indica el número de fila dentro del rango. Si cero, devuelve todas las filas.
- **Número_columna** es un argumento opcional que indica el número de columna dentro del rango. Si cero, devuelve todas las columnas.
- **Número_área** es un argumento opcional que indica el número de área si el **Rango** es múltiple.

En el siguiente ejemplo, se muestra el contenido de la segunda fila y tercera columna del rango de datos.

Formulación avanzada en LibreOffice Calc

	A	B	C	D	E	F	G	H	I
21	DATOS								
22	Id	Nombre	Ciudad	Importe					
23	1	Joaquín	Alicante	100	Fila:		2		
24	3	Carmen	Valencia	200	Columna:		3		
25	5	Pepito	Barcelona	300	Fórmula:		=INDICE(\$A\$23:\$D\$27;\$G\$23;\$G\$24)		
26	7	Fulano	Madrid	400	Resultado:		Valencia		
27	9	Mengana	Reus	500					

Si el rango es una sola fila o columna, el uso es trivial. En el ejemplo siguiente, se muestra el cuarto elemento de la lista de nombres.

	A	B	C	
21	DATOS			
22	Id	Nombre	Ciudad	Im
23	1	Joaquín	Alicante	
24	3	Carmen	Valencia	
25	5	Pepito	Barcelona	
26	7	Fulano	Madrid	
27	9	Mengana	Reus	
28				
29	Fórmula	=INDICE(\$B\$23:\$B\$27;4)		
30	Resultado:	Fulano		
31				

Un uso más sofisticado, dado que la función **ÍNDICE** realmente devuelve una referencia, sería calcular el promedio de la cuarta columna del rango.

	A	B	C	D	E
21	DATOS				
22	Id	Nombre	Ciudad	Importe	
23	1	Joaquín	Alicante	100	
24	3	Carmen	Valencia	200	
25	5	Pepito	Barcelona	300	
26	7	Fulano	Madrid	400	
27	9	Mengana	Reus	500	
28					
29	Fórmula	=PROMEDIO(INDICE(\$A\$23:\$D\$27;0;4))			
30	Resultado:	300			
31					

Función COINCIDIR()

A diferencia de la anterior, **COINCIDIR** nos informa de la posición de un dato dentro de un rango (fila o columna).

COINCIDIR devuelve la posición relativa dentro de un rango formado por una sola fila o columna.

Sintaxis básica:

=COINCIDIR(Dato_buscado;Rango;Tipo)

donde

- **Dato_buscado** es el dato que se busca. Admite expresiones regulares si el **Tipo** es cero.
- **Rango** es la fila o columna donde se desea encontrar el **Dato_buscado**.
- **Tipo** es un argumento opcional que especifica el tipo de búsqueda:

Formulación avanzada en LibreOffice Calc

- Omitido o 1: se considera que **Rango** está ordenado ascendentemente y la búsqueda será aproximada, devolviendo la posición del mayor elemento menor o igual al buscado.
- 0: efectuará una búsqueda exacta, admitiendo expresiones regulares en el **Dato_buscado**.
- -1: asume que **Rango** está ordenado descendientemente y la búsqueda será aproximada, devolviendo la posición del último elemento mayor o igual al buscado.

En el siguiente ejemplo se busca la posición de **Carmen** dentro de la lista de nombres. Obsérvese el uso de expresiones regulares en el dato buscado.

	A	B	C	D	E
21	DATOS				
22	Id	Nombre	Ciudad	Importe	
23	1	Joaquín	Alicante	100	
24	3	Carmen	Valencia	200	
25	5	Pepito	Barcelona	300	
26	7	Fulano	Madrid	400	
27	9	Mengana	Reus	500	
28					
29	Dato buscado:	Car.*			
30	Fórmula:	=COINCIDIR(\$B\$29;\$B\$23:\$B\$27;0)			
31	Resultado:	2			
32					

Ahora, combinando el uso de **COINCIDIR** y **ÍNDICE** podemos superar la limitación de **BUSCARV** que restringe la búsqueda a la primera columna.

En el ejemplo, buscamos el **Nombre Pep.*** y obtenemos el **Id** correspondiente.

	A	B	C	D	E	F
21	DATOS					
22	Id	Nombre	Ciudad	Importe		
23	1	Joaquín	Alicante	100		
24	3	Carmen	Valencia	200		
25	5	Pepito	Barcelona	300		
26	7	Fulano	Madrid	400		
27	9	Mengana	Reus	500		
28						
29	Nombre:	Pep.*				
30	Fórmula:	=ÍNDICE(\$A\$23:\$A\$27;COINCIDIR(\$B\$29;\$B\$23:\$B\$27;0))				
31	Resultado:	5				
32						

Mediante **COINCIDIR** se encuentra la posición relativa del nombre buscado dentro de la lista de nombres, y el resultado es el argumento que se pasa a **ÍNDICE** para que devuelva el **Id** correspondiente.

Hemos visto la "magia" que podemos hacer combinando las diferentes funciones que **Calc** pone a nuestra disposición.

Una vez más recomendamos una exploración en profundidad de otras funciones que no hemos tratado en estos apartados.

Todo el tiempo invertido en ello, será revertido con creces cuando lo apliquemos a nuestro trabajo.

Practica lo aprendido

Para practicar los temas vistos en esta unidad didáctica te sugerimos los siguientes ejercicios. Prepárate

Formulación avanzada en LibreOffice Calc

para invertir unas cuantas horas en ellos, pero te aseguramos que, tras su resolución, podrás afirmar que ya dominas la formulación en hojas de cálculo. ;-) (guiño)

Los archivos de prácticas correspondientes se adjuntan en el propio enunciado.

¡Disfruta de la tarea!



Ejercicio para practicar el uso del Asistente

Vamos a explorar el uso del asistente, mediante un sencillo problema de cálculo de préstamos.

Realiza las tareas siguientes:

1. Crea una nueva hoja de cálculo y guarda el documento con el nombre **calculo-prestamos.ods**.
2. Reproduce el contenido que se muestra en la imagen

	A	B	C
1	Cálculo de préstamos		
2			
3	Capital:	200.000 €	
4	Tipo interés:	4,00% anual	
5	Plazo:	30 años	
6			
7	Cuota:		mensual
8			
9			

3. En la celda **B7** deberás introducir la fórmula que te permita calcular la cuota mensual de un préstamo de las características introducidas. Para ello usarás el **Asistente de funciones** y la **Ayuda** del programa en la medida que la necesites. Pistas:
 - La función que se necesita es **PAGO**.
 - Ten presente que el tipo de interés y el plazo está dado en años y lo que deseamos obtener son cuotas mensuales.
 - El resultado correcto es de **-954,83 €** (en negativo, pues las funciones financieras diferencian los cobros de los pagos por el signo. Si has cobrado "en positivo", el pago para devolverlo es "en negativo"
4. Guarda los cambios.



Ejercicio de CONTAR.SI y SUMAR.SI

Para la realización de este ejercicio, necesitarás el archivo que se enlaza a continuación:

■ [prehistoric-computer.ods](#)

Realiza las siguientes tareas:


1. Abre el archivo **prehistoric-computer.ods**. Esta hoja de cálculo contiene una relación de modelos de ordenador con las unidades disponibles de cada modelo.
2. Debes completar la tabla de resumen (rango **D3:F7**) con las fórmulas apropiadas para que se muestren la **Cantidad** total de **Modelos** diferentes hay de cada **Serie** y la **Suma Uds.**

Formulación avanzada en LibreOffice Calc


de cada serie, así como el **Total** de ambos conceptos. Para tu comprobación, en la siguiente imagen se muestra el resultado:

Serie	Cantidad	Suma Uds.
DX.*	5	79
SX.*	3	41
LB.*	5	84
Total	13	204

3. Crea un gráfico de barras como el siguiente que muestre la cantidad de unidades de cada modelo



4. Crea un gráfico circular que muestre la suma de unidades por cada serie.



5. Guarda los cambios.



Ejercicio de cálculo con fechas

Para la realización de esta práctica necesitarás el archivo que puedes descargar del siguiente

enlace:

■ [antiguedad-saldos.ods](#)

Realiza las siguientes tareas:

1. Abre el archivo **antiguedad-saldos.ods** y guárdalo con el nombre **antiguedad-calculada.ods**.
2. Establece en la celda **E1** el valor 15/07/07.
3. En la celda **B15** introduce la suma del **Importe** (rango **C4:C11**) de las facturas pendientes y aplícale formato Euro.
4. Calcula el valor de **Retraso días** (rango **E4:E11**) como la diferencia entre la fecha de la celda **E1** y el **Vencimiento** (rango **D4:D11**) de cada factura.
5. Para calcular el **Retraso medio ponderado** de la totalidad de las facturas (similar al sistema que usan los bancos para calcular el saldo medio), efectúa los siguientes pasos:
 - Añade en el rango **F4:F11** el producto de multiplicar el **Retraso días** de cada factura por su **Importe**.
 - En la celda **F12** introduce la suma de los valores del rango **F4:F11**.
 - En la celda **B16** divide el valor de la celda **F12** entre el valor de la suma de las facturas **B15**.
 - Introduce en la celda **E12** el promedio de **Retraso días** y comprueba que el retraso medio ponderado es diferente del simple promedio de los días de retraso de cada factura, pues cada retraso tiene un “peso” diferente en función del importe al que afecta.
 - Asimismo, aprovecha para aprender el manejo de la función **SUMA.PRODUCTO** que te permitiría obtener directamente el valor de la celda **F12** sin necesidad de cálculos intermedios.
6. Guarda los cambios.

El resultado final es como se muestra en la imagen. Fíjate en la barra de fórmulas, donde se ve el uso de **SUMA.PRODUCTO**.

F14	=SUMA.PRODUCTO(C4:C11;E4:E11)					
	A	B	C	D	E	F
1	Relación de facturas pendientes de cobro			a fecha:	15/07/2007	
2						
3	Cliente	Factura	Importe	Vencimiento	Retraso días	
4	Carnicer Heras	197	9908	14/06/2007	31	307148
5	Alvarez Justo	188	11779	16/04/2007	90	1060110
6	Arana Higuera	205	7719	23/06/2007	22	169818
7	Hidalgo Jimena	197	11412	04/04/2007	102	1164024
8	Lopez Vara	157	639	24/02/2007	141	90099
9	Martín Peña	204	1064	06/05/2007	70	74480
10	Sierra Garzón	229	1436	23/04/2007	83	119188
11	González Suevo	205	2972	24/06/2007	21	62412
12				Promedio:	70	3047279
13						
14				usando SUMAPRODUCTO:		3047279
15	Total importe	46.929,00 €				
16	Retraso medio ponderado	64,93				
17						



Ejercicios de funciones de texto

Para realizar estas prácticas necesitarás los archivos que se enlazan a continuación:

- [contactos.ods](#)
- [reordena-nombres.ods](#)

Realiza las siguientes tareas:

1. Abre el libro **contactos.ods** y guárdalo con el nombre **contactos-nombre-completo.ods**.
2. Añade una nueva hoja y nómbrala como **NombreCompleto**.
3. En la celda **A1** de la hoja **NombreCompleto** escribe el texto: **Nombre completo** y ponlo en negrita.
4. En la celda **A2** escribe una fórmula que muestre en formato de nombre propio (primera letra de cada palabra en mayúsculas y resto de letras en minúsculas) el nombre completo (nombre y apellidos separados por un espacio) del primer contacto de la **Hoja1**.
5. Copia la fórmula de la celda **A2** en todo el rango **A2:A123**. El resultado será como se muestra en la imagen.

	A	B
1	Nombre completo	
2	Estefania Arocas Pasadas	
3	Queralt Viso Gilabert	
4	Joan Ayala Ferreras	
5	Joan Baez Tejado	
6	Marc Bastardes Soto	
7	Josep Anguera Vilafranca	
8	Esther Pascual Aloy	
9	Laura Vallés Girvent	
10	Raquel Rava Garcia	

6. Guarda los cambios y cierra el archivo.
7. Abre el documento **reordena-nombres.ods**.
8. El documento consiste en una relación de nombres en formato **APELLIDOS, NOMBRE**. Se trata de transformarlo en una lista en formato "humano": **Nombre Apellidos**. Para facilitar la tarea se han previsto las columnas para los pasos intermedios que se necesitarán y que son:
 - Longitud
 - Posición coma
 - Apellido
 - Nombre

■ Sobre todas ellas hay una pequeña ayuda en forma de comentario en el propio documento que podrás ver pasando el cursor sobre las celdas de título con una pequeña marca roja.
9. Una vez resuelto, su apariencia será como se muestra en la imagen

Formulación avanzada en LibreOffice Calc

	A	B	C	D	E	F
1	Apellido, Nombre	Longitud	Posición coma	Apellido	Nombre	Nombre Apellido
2	BASTARDAS FRANCH, ANA INÉS	26	17	BASTARDAS	ANA INÉS	Ana Inés Bastardas Franch
3	ABADIAS MASANA, IVET	20	15	ABADIAS M	IVET	Ivet Abadías Masana
4	AREVALO SANCHEZ, JULIA	22	16	AREVALO S	JULIA	Júlia Arevalo Sanchez
5	ALINS MULET, DANIEL	19	12	ALINS MULET	DANIEL	Daniel Alins Mulet
6	GARCIA GONZÁLEZ, ABEL	21	16	GARCIA GON	ABEL	Abel García González
7	ALVAREZ PARCERISA, IRENE	24	18	ALVAREZ PA	IRENE	Irene Alvarez Parcerisa
8	CASAS ANDRÉS, ADRIÀ	19	13	CASAS AND	ADRIÀ	Adrià Casas Andrés
9	MORALES GESE, JAIRO	19	13	MORALES G	JAIRO	Jairo Morales Gese
10	BARALDÉS MARTORELL, CRISTINA	28	19	BARALDÉS	CRISTINA	Cristina Baraldés Martorell
11	AROCA GÓMEZ, DAVID	18	12	AROCA GOM	DAVID	David Aroca Gómez
12	RUEDA ALVAREZ, ADRIÀ	20	14	RUEDA ALV	ADRIÀ	Adrià Rueda Alvarez
13	ALVAREZ DOMENECH, LUCIA	23	17	ALVAREZ DO	LUCIA	Lucia Alvarez Domenech
14	BOIX GONZÁLEZ, CARLA	20	14	BOIX GONZA	CARLA	Carla Boix González
15	BARALDÉS MONROS, ADRIÀ	22	16	BARALDÉS	ADRIÀ	Adrià Baraldés Monrós
16	AGUILERA MERINO, MARTA	22	16	AGUILERA M	MARTA	Marta Aguilera Merino
17	BAREA D'HAENE, MARC	19	14	BAREA D'HA	MARC	Marc Barea D'Haene
18	BARROSO D'HAENE, ALEX	21	16	BARROSO D	ALEX	Alex Barroso D'Haene

- Si vas "para nota", tengo un reto para ti: haz la conversión en una sola fórmula, anidando funciones y sin necesidad de columnas intermedias. ;-)
- Guarda los cambios.



Ejercicios de funcion SI

Para la realización de estas prácticas necesitarás los archivos enlazados seguidamente:

- [calificaciones.ods](#)
- [vendedores-eficaces.ods](#)
- [vencimiento-clientes.ods](#)

Realiza las siguientes tareas:

- Abre el documento **calificaciones.ods** y guárdalo con el nombre **calificaciones-simples.ods**. Esta hoja de cálculo muestra las notas trimestrales de un grupo de alumnos.
- Calcula la **Nota** de cada alumno como promedio de sus notas trimestrales.
- En la columna **Evaluación** escribe la fórmula que muestre el texto **APTO** para los alumnos cuya **Nota** iguale o supere a 5 y **NO APTO** en caso contrario. El resultado será como se muestra en la imagen.

	A	B	C	D	E	F
1	NOTAS ALUMNOS					
2						
3	Nombre alumno	Trim 1	Trim 2	Trim 3	Nota	Evaluación
4	Mariano Pérez	2	3	5	3,33	NO APTO
5	Susana Díaz	9	8	4	7,00	APTO
6	Margarita Soler	6	6	5,75	5,92	APTO
7	Montse Abril	7	4	5	5,33	APTO
8	Joan Cisa	4	8	6	6,00	APTO
9	Jose Sánchez	6	6,5	8	6,83	APTO
10	Albert Català	5	4	9	6,00	APTO
11	Robert Amic	9,1	7	4	6,70	APTO
12	Anna Torres	1	6	2,3	3,10	NO APTO
13	Javier García	9	10	9	9,33	APTO

- Guarda los cambios.
- Abre el documento **vendedores-eficaces.ods** y guárdalo con el nombre **vendedores-eficaces-calculados.ods**.

Formulación avanzada en LibreOffice Calc

Esta hoja de muestra una lista de teleoperadores que trabajan diferentes **Zonas** de ventas y a partir de las **Llamadas** efectuadas, conciertan diferentes **Visitas**. Se pretende medir la efectividad de los diferentes vendedores.

- Calcular la **Eficacia** de cada vendedor (rango **E4:E13**) como el porcentaje de **Visitas** sobre las **Llamadas** efectuadas.
- En la columna **Sí/No**, establecer **Sí** cuando la **Eficacia** del vendedor sea mayor o igual al **Umbral de eficacia** establecido en la celda **C1**. En caso contrario se mostrará **No**.
- En las celdas **F14** y **F15**, calcular respectivamente el total de vendedores considerados eficaces y el total de los no eficaces.

Pista: usar **CONTAR.SI**. El resultado será como muestra la imagen:

	A	B	C	D	E	F
1	Umbral de eficacia:		40,00%			
2						
3	Vendedor	Zona	Llamadas	Visitas	Eficacia	Sí/No
4	López	sur	7	1	14,3%	No
5	Dieguez	norte	9	1	11,1%	No
6	López	oeste	12	2	16,7%	No
7	Dieguez	sur	15	4	26,7%	No
8	Sastre	sur	16	8	50,0%	Sí
9	Dieguez	este	19	6	31,6%	No
10	Sastre	oeste	21	9	42,9%	Sí
11	Sastre	este	25	10	40,0%	Sí
12	López	norte	34	15	44,1%	Sí
13	Sastre	norte	60	13	21,7%	No
14					Eficaces	4
15					No eficaces	6
16						

- Añade un resumen por zonas que nos permita obtener el total de **Llamadas**, **Visitas** y **Eficacia** de cada Zona.

Pista: usar **SUMAR.SI**. Si escribes las referencias mixtas adecuadas, la suma de llamadas y visitas la puedes obtener con una sola fórmula. La apariencia final debería ser como se muestra en la imagen:

17						
18		Zona	Llamadas	Visitas	Eficacia	Sí/No
19		norte	103	29	28,2%	No
20		sur	38	13	34,2%	No
21		este	44	16	36,4%	No
22		oeste	33	11	33,3%	No
23						

- Guarda los cambios.
- Abre el archivo **vencimiento-clientes.ods** y guárdalo con el nombre **vencimiento-clientes-calculado.ods**. El documento tiene la siguiente apariencia.

	A	B	C	D	E	F	G	H
1	Cálculo vencimientos facturas clientes							
2								
3	Fecha factura	Plazo pago	Día de pago	Vencimiento teórico	Día teórico	Mes teórico	Mes real	Vencimiento real
4	01/04/2013	90	20					
5	13/04/2013	0	5					
6	14/04/2013	60	20					
7	22/04/2013	30	15					
8	24/04/2013	15	10					
9	28/04/2013	30	10					
10	28/04/2013	60	25					
11								

- Este modelo permite simular un problema típico de los sistemas de facturación, en el cual

Formulación avanzada en LibreOffice Calc

se establece un plazo en días (a 30, 60 ó 90 días), pero el cliente tiene un día de pago, que provocará un “corrimiento” de un mes, en el caso de que el día del vencimiento “teórico” sea posterior al día de pago establecido por este cliente. Para resolverlo se usarán diferentes funciones de tratamiento de fechas y la función lógica **SI**. En los pasos siguientes se detalla como realizar el cálculo:

- El **Vencimiento teórico** se obtiene incrementando la **Fecha factura** en el número de días establecido en **Plazo pago**.
- El **Día teórico** es el número del día correspondiente al **Vencimiento teórico**.
- El **Mes teórico** es el número de mes correspondiente al **Vencimiento teórico**.
- Si el **Día teórico** es posterior al **Día de pago**, el **Mes real** será igual al **Mes teórico** incrementado en 1, de lo contrario, el **Mes real** corresponderá al mismo **Mes teórico**.
- El **Vencimiento real** se construirá con la fecha obtenida a partir del **Día de pago**, el **Mes real** y el año correspondiente al **Vencimiento teórico**.
- Aunque en el modelo presentado no se da el caso, ten presente que si la fecha resultase algo tan incoherente como 15/13/2013, el propio **Calc** convertiría la fecha en 15/01/2014. En la imagen siguiente puedes ver el resultado.

	A	B	C	D	E	F	G	H
1	Cálculo vencimientos facturas clientes							
2								
3	Fecha factura	Plazo pago	Día de pago	Vencimiento teórico	Día teórico	Mes teórico	Mes real	Vencimiento real
4	01/04/2013	90	20	30/06/13	30	6	7	20/07/13
5	13/04/2013	0	5	13/04/13	13	4	5	05/05/13
6	14/04/2013	60	20	13/06/13	13	6	6	20/06/13
7	22/04/2013	30	15	22/05/13	22	5	6	15/06/13
8	24/04/2013	15	10	09/05/13	9	5	5	10/05/13
9	28/04/2013	30	10	28/05/13	28	5	6	10/06/13
10	28/04/2013	60	25	27/06/13	27	6	7	25/07/13
11								
12								

13. Con finalidad didáctica se han establecido unas columnas para obtener los pasos intermedios, pero el objetivo para demostrar tu dominio de la formulación avanzada sería crear la fórmula para obtener el **Vencimiento real** exclusivamente con los datos de **Fecha factura**, **Plazo pago** y **Día de pago**. Si vas "para nota", resuélvelo así. ;-)
14. Guarda los cambios en el documento.



Ejercicios de función BUSCARV

Para la realización de estas prácticas necesitarás el archivo **calificaciones.ods** que ya descargaste en un ejercicio anterior. También necesitarás los dos documentos que se enlazan seguidamente:

- [cambio-formulario.ods](#)
- [cambio-datos.ods](#)

Realiza las siguientes tareas:

1. Abre el documento **calificaciones.ods** y guárdalo con el nombre **calificaciones-complejas.ods**. Este libro muestra las notas trimestrales de un grupo de alumnos.
2. Calcula la **Nota** de cada alumno como promedio de sus notas trimestrales.
3. Para establecer la **Evaluación** de cada alumno se hará con los conceptos mostrados en el rango **A22 : A25**. Para añadir las notas de corte para cada evaluación inserta celdas vacías

Formulación avanzada en LibreOffice Calc

en el rango **A21:A25** y complétalas con los valores que se muestran en la imagen.

	A	B	C	D
20				
21			Número	%
22	0	Insuficiente		
23	5	Aprobado		
24	7	Notable		
25	9	Excelente		
26				
27				

4. En la columna **Evaluación** (**F4:F13**) escribe la fórmula que muestre la evaluación correspondiente a la nota de cada alumno en función de estos tramos. Para facilitar la escritura de la fórmula, se recomienda dar nombre al rango **A22:B25**, y usar este nombre como argumento en la función **BUSCARV** que utilizarás para resolverlo. El resultado debe ser como sigue.

	A	B	C	D	E	F
1	NOTAS ALUMNOS					
2						
3	Nombre alumno	Trim 1	Trim 2	Trim 3	Nota	Evaluación
4	Mariano Pérez	2	3	5	3,33	Insuficiente
5	Susana Díaz	9	8	4	7,00	Notable
6	Margarita Soler	6	6	5,75	5,92	Aprobado
7	Montse Abril	7	4	5	5,33	Aprobado
8	Joan Cisa	4	8	6	6,00	Aprobado
9	Jose Sánchez	6	6,5	8	6,83	Aprobado
10	Albert Català	5	4	9	6,00	Aprobado
11	Robert Amic	9,1	7	4	6,70	Aprobado
12	Anna Torres	1	6	2,3	3,10	Insuficiente
13	Javier García	9	10	9	9,33	Excelente
14						

5. Aprovecha para completar la hoja con el resto de fórmulas que se piden en la misma. **Nota:** para el cálculo de la nota más repetida, un comentario en la celda recomienda usar la función **MODA**; parece que por un error de traducción en algunas versiones está disponible con el nombre de **MODO**.
6. Guarda los cambios.
7. Abre el archivo **cambio-formulario.ods** y guárdalo con el nombre **cambio-formulario-completado.ods**.
8. Abre el libro **cambio-datos.ods** y dispón las ventanas para ver ambos documentos en paralelo.

The screenshot shows two LibreOffice Calc windows. The left window, titled 'cambio-formulario-completado.ods', displays a spreadsheet with columns A (Fecha), B (Dólares), C (Cambio), and D (Total). The data for the first four rows is as follows:

Fecha	Dólares	Cambio	Total
23-jun-13	20		
01-jul-13	2		
07-jul-13	5		
29-jun-13	10		

The right window, titled 'cambio-datos.ods', displays a spreadsheet with columns A (Día) and B (Dólar). The data for the first 15 rows is as follows:

Día	Dólar
21/06/13	1.200,00
22/06/13	1.212,00
23/06/13	1.224,12
24/06/13	1.236,36
25/06/13	1.248,72
26/06/13	1.261,21
27/06/13	1.273,82
28/06/13	1.286,56
29/06/13	1.299,43
30/06/13	1.312,42
01/07/13	1.325,55
02/07/13	1.338,80
03/07/13	1.352,19
04/07/13	1.365,71
05/07/13	1.379,37

9. Escribe una fórmula en la columna **Cambio** (**C5:C8**) del formulario que localice en la tabla

Formulación avanzada en LibreOffice Calc

de datos el cambio correspondiente a cada uno de los días mostrados en el rango **A5 : A8**.

Pista: La búsqueda debe ser exacta.

Nota: Al señalar las celdas del rango donde buscar en otro documento, no funciona fijar la referencia con **Mayúsc+F4**; deberás añadir a mano las notaciones con símbolo \$ para fijar la referencia.

10. Para completar el ejercicio, calcula el **Total** (D5 : D8) como producto de **Dólares** por **Cambio** y en **Importe total** (D10) suma los importes de **Total**. La apariencia del formulario será como se muestra:

	A	B	C	D	
1	<u>Cambios del dólar</u>				
2					
3	Fecha	Dólares	Cambio	Total	
4					
5	23-jun-13	20	1.224,12	24.482,40	
6	01-jul-13	2	1.325,55	2.651,10	
7	07-jul-13	5	1.406,09	7.030,45	
8	29-jun-13	10	1.299,43	12.994,30	
9					
10			Importe total	47.158,25	
11					

11. Cierra el documento **cambio-datos.ods**.
12. Guarda los cambios en **cambio-formulario.ods**.
13. Cambia las fechas del formulario (dentro de las fechas comprendidas en la tabla de datos) y comprueba que sigue funcionando perfectamente aunque la tabla de datos está cerrada. Introduce una fecha fuera del intervalo de estas fechas y comprueba que aparece el error **#N/D**.
14. Cierra el documento sin guardar los últimos cambios.



Ejercicio de referencias mixtas

Para la realización de este ejercicio, necesitarás el archivo que puedes descargar desde el enlace siguiente:

■ [calculo-precios.ods](#)

En este ejercicio debes calcular en una hoja de cálculo el precio unitario de un determinado artículo en función del descuento por volumen de compra y de la fecha de pago. Realiza las siguientes tareas.

1. Abre el archivo **calculo-precios.ods** que te mostrará la siguiente hoja de cálculo

Formulación avanzada en LibreOffice Calc


	A	B	C	D	E	F	G
1							
2		Cantidad	Descuento	Contado	30 días	60 días	90 días
3		1000	0%				
4		5000	2%				
5		10000	5%				
6		15000	10%				
7		30000	15%				
8							
9			Recargos	-5%	0%	5%	7%
10							
11		Precio sin descuento		2.000,00 €			
12							

2. Hay que completar el rango D3:G7 con una fórmula que determine el **Precio unitario** de un producto en función de su **Precio sin descuento**, un porcentaje de **Descuento** y un porcentaje de **Recargos**., con las siguientes consideraciones:

- una posible expresión de la fórmula (hay otras) sería: **= Precio sin descuento – (Descuento * Precio sin descuento) + (Recargos * Precio sin descuento)**
- Puedes intentar resolverlo mediante referencias mixtas, pero este es un ejemplo en el cual el uso de nombres de rangos te resultará mucho más fácil.
- El rango B3:B7 que indica la **Cantidad** no es relevante para el cálculo, simplemente permite que el usuario pueda saber que descuento aplicar a un pedido en concreto.
- Una vez resuelto, la hoja deberá mostrar la siguiente información:

	A	B	C	D	E	F	G
1							
2		Cantidad	Descuento	Contado	30 días	60 días	90 días
3		1000	0%	1.900,00 €	2.000,00 €	2.100,00 €	2.140,00 €
4		5000	2%	1.860,00 €	1.960,00 €	2.060,00 €	2.100,00 €
5		10000	5%	1.800,00 €	1.900,00 €	2.000,00 €	2.040,00 €
6		15000	10%	1.700,00 €	1.800,00 €	1.900,00 €	1.940,00 €
7		30000	15%	1.600,00 €	1.700,00 €	1.800,00 €	1.840,00 €
8							
9			Recargos	-5%	0%	5%	7%
10							
11		Precio sin descuento		2.000,00 €			
12							

3. Crea un gráfico que muestre la información de los diversos precios resultantes, con un aspecto similar al de la imagen



4. Guarda los cambios.

